

Débuter avec Cast3M

calculs thermo mécaniques

François Di Paola


CEA Saclay,
<https://www-cast3m.cea.fr>

Janvier 2025

Sommaire

- Présentation de Cast3M
- Le langage Gibiane
- Travaux dirigés

Thermo mécanique d'une structure avec un trou



- Compléments

Cast3M, quid ?

Logiciel de simulation utilisant la **méthode des éléments finis** en
mécanique/thermique des **structures** et des **fluides**

Cast3M, quid ?




Logiciel de simulation utilisant la **méthode des éléments finis** en **mécanique/thermique** des **structures** et des **fluides**

- Résolution d'**équations aux dérivées partielles**

Cast3M, quid ?

Logiciel de simulation utilisant la **méthode des éléments finis** en **mécanique/thermique** des **structures** et des **fluides**




- Résolution d'**équations aux dérivées partielles**
- **Système complet** : solveur, pré/post-processeur, visualisation, import/export des données. . .



Cast3M, quid ?

Logiciel de simulation utilisant la **méthode des éléments finis** en **mécanique/thermique** des **structures** et des **fluides**

- Résolution d'**équations aux dérivées partielles**
- **Système complet** : solveur, pré/post-processeur, visualisation, import/export des données...




- Basé sur un langage de commande : **Gibiane** (orienté objet)


Nombreux domaines d'application

- **Mécanique des structures**


Quasi-statique (non linéarités matériau, géométrie, conditions limites)



Microstructure : agrégat polycristallin



Composant : liaison d'un tube




Batiment en béton armé (S. Durand)

Nombreux domaines d'application

- **Mécanique des structures**

Quasi-statique (non linéarités matériau, géométrie, conditions limites)

Contact/frottement, Flambage



Attachement d'un clip (contact + plasticité)


Nombreux domaines d'application

- **Mécanique des structures**

Quasi-statique (non linéarités matériau, géométrie, conditions limites)

Contact/frottement, **Flambage**

Dynamique (temporelle, modale, interaction fluide structure)



Premiers modes propres d'une cymbale

Nombreux domaines d'application


- **Mécanique des structures**

Quasi-statique (non linéarités matériau, géométrie, conditions limites)

Contact/frottement, **Flambage**

Dynamique (temporelle, modale, interaction fluide structure)

Rupture (XFEM, propagation dynamique, zones cohésives)



Rupture d'éprouvette CT, plasticité/endommagement, suppression d'éléments lors du calcul (S. Kebiri)

Nombreux domaines d'application

- **Mécanique des structures**

Quasi-statique (non linéarités matériau, géométrie, conditions limites)


Contact/frottement, Flambage

Dynamique (temporelle, modale, interaction fluide structure)

Rupture (XFEM, propagation dynamique, zones cohésives)

- **Thermique**

Conduction, convection, advection, rayonnement, changement de phase



Nombreux domaines d'application

- **Mécanique des structures**

Quasi-statique (non linéarités matériau, géométrie, conditions limites)

Contact/frottement, Flambage

Dynamique (temporelle, modale, interaction fluide structure)

Rupture (XFEM, propagation dynamique, zones cohésives)


- **Thermique**

Conduction, convection, advection, rayonnement, changement de phase

- **Mécanique des fluides**

- **Diffusion** multi espèces (loi de Fick)

- Fabrication additive, Métallurgie



Proportion de bainite lors d'une fabrication additive (C. Berthier)

Nombreux domaines d'application

- **Mécanique des structures**

Quasi-statique (non linéarités matériau, géométrie, conditions limites)

Contact/frottement, Flambage

Dynamique (temporelle, modale, interaction fluide structure)

Rupture (XFEM, propagation dynamique, zones cohésives)


- **Thermique**

Conduction, convection, advection, rayonnement, changement de phase

- **Mécanique des fluides**

- **Diffusion** multi espèces (loi de Fick)

- Fabrication additive, Métallurgie



Simulation magnéto thermo hydrodynamique du soudage TIG (arc plasma + bain) (C. Nahed)

Nombreux domaines d'application

- **Mécanique des structures**

Quasi-statique (non linéarités matériau, géométrie, conditions limites)

Contact/frottement, Flambage

Dynamique (temporelle, modale, interaction fluide structure)

Rupture (XFEM, propagation dynamique, zones cohésives)

- **Thermique**

Conduction, convection, advection, rayonnement, changement de phase

- **Mécanique des fluides**


- **Diffusion** multi espèces (loi de Fick)

- Fabrication additive, Métallurgie

- Magnétostatique

- Couplage thermo-hygro-mécanique

- Optimisation topologique



Optimisation topologique d'un pont

SCAL
< 1.00E+00
> 1.01E-03


1.0
0.95
0.90
0.85
0.80
0.75
0.70
0.65
0.60
0.55
0.50
0.45
0.40
0.35
0.30
0.25
0.20
0.15
0.10
5.00E-02
0.0

Comment obtenir Cast3M ?

- Multi plateformes
Windows, Linux, macOS
- Où télécharger Cast3M ?
<https://www-cast3m.cea.fr/index.php?page=dlcastem>
- Accès au code source
Développement communautaire
Compilateur / éditeur de liens fournis
- Prix
Gratuit pour la recherche et l'enseignement
Payant pour une utilisation commerciale
- Utilisateurs/clients
Universités, écoles d'ingénieurs
IRSN, EDF, SNCF, CNRS, Framatome, Air Liquide, CERN, ...
Outil de référence IRSN pour les analyse de sureté des installations nucléaires françaises
Outil de référence Framatome pour l'analyse en mécanique de la rupture


Comment utiliser Cast3M ?

1. Écrire un fichier texte en Gibiane
et l'enregistrer dans un répertoire de travail



Comment utiliser Cast3M ?

1. Écrire un fichier texte en Gibiane
et l'enregistrer dans un répertoire de travail
2. Ouvrir un terminal / invite de commandes
et se placer dans le répertoire
3. Lancer Cast3M sur ce fichier
castem24 hello.dgibi
4. Utilisable aussi sans fichier
mode interactif
castem24



```
Selection Cast3M 2023 - 64bits : hello.dgibi
le repertoire ./procedur
le repertoire d'installation

*****
$ "castem24 hello world" ;
$ "FIN ."
***** Arret du programme Cast3M. Niveau d'erreur: 0 *****

INFORMATIONS FINALES
-----
$ DATE DEBUT   : 07/06/2024      HEURE DEBUT : 20:01:56,92
$ DATE FIN     : 07/06/2024      HEURE FIN  : 20:03:57,22
$              :                  DUREE      : 00:00:00,30
$
$ Support Cast3M : http://www-cast3m cea.fr/index.php/page-mailsupport
$ Site Web Cast3M : http://www-cast3m cea.fr/index.php
C:\Users\F0218221\Desktop
```

Le site web Cast3M

<https://www-cast3m.cea.fr/>

- Présentation de Cast3M
- Formation et tutoriels vidéo
- Documentation (notices, manuels, sources, exemples)
- Fiches d'anomalie et de développement
- Téléchargements
- Contact : support Cast3M
- Communauté : liste de diffusion, club Cast3M

Le langage Gibiane

- **Langage de programmation** destiné au calcul EF
Objets classiques (entiers, flottants, chaînes, logiques, tables)
Instructions conditionnelles
Boucles itératives
Sous structuration
Récursivité
- **Langage interprété**
Le programme peut être exécuté dès que le script est modifié
Le programme peut être exécuté en mode interactif
- **Langage orienté objet**
Tout est traité comme un objet
Pas besoin de déclarer les variables ou de spécifier leur type
- Mots clefs en français
- Programmation facile et rapide



Gibiane : règles de syntaxe

- Ligne(s) de commande
500 caractères max par instruction
Une instruction peut être écrite sur plusieurs lignes
Se termine par un point virgule ;
Le symbole d'affectation est le signe égal =
- Insensibilité à la casse
 $TOTO = 3.14$;
 $A = 2. * tOTO$; la variable A vaut bien 6.28
- Fin du fichier de données
Commande FIN ; $\Rightarrow$ arrêt de Cast3M
Ligne vide ou un EOF $\Rightarrow$ mode interactif
- Ligne de commentaire : commence par *
- Lignes vides autorisées

Gibiane : règles de syntaxe

- Pas de priorité des opérations (lecture de gauche à droite)
 $1+2*3 = 9$
 $1+(2*3) = 7$
- Quelques **interdictions** :
Pas de **tabulations** $\Rightarrow$ messages d'erreur incompréhensibles
Pas de **double quotes** "
- Quelques (fortes !) **recommandations** :
Pas de **caractères spéciaux** (é, ç, œ, ...)
Utiliser une **indentation** (avec des espace)
Régler son éditeur de texte : coloration syntaxique, remplacement des tabulations par des espaces
- Quelques **erreurs** classiques :
Point virgule ; oublié en fin de ligne $\Rightarrow$ la lecture de l'instruction continue !
Apostrophe ' oubliée à la fin d'une chaîne $\Rightarrow$ la définition de la chaîne continue !

Gibiane : les objets

- Définition
Désigne toute **structure de données/résultats** munie d'un **type**
(éventuellement d'un sous-type) et d'un **nom**
- Noms des objets
Donné par l'utilisateur
Limité à 24 caractères parmi : a...z A...Z 0...9 _
Pièges classiques :
 plus de 24 caractères : les surnuméraires sont ignorés
 utilisation du tiret – ⇒ interdit !
 caractères accentués é, è ⇒ interdit !
- Type des objets
Il existe plus de 40 types d'objets différents

Gibiane : les objets

- Exemple

```
obj1  = 3 ;  
obj2  = 3.14 ;  
obj3  = MOT 'Hello' ;  
obj4  = vrai ;  
poin1 = POINT 0. 0. ;  
poin2 = POINT 28. 3. ;  
obj5  = DROITE 5 poin1 poin2 ;
```

```
LISTER obj5 ;
```

```
MAILLAGE 3520406 : 5 element(s) de type SEG2
```

```
0 sous-reference(s)
```

```
1ere ligne  numero element : 2eme couleur : 3eme... noeud(s)
```

1	2	3	4	5
DEFA	DEFA	DEFA	DEFA	DEFA
1	3	4	5	6
3	4	5	6	2

Gibiane : les opérateurs

- Définition

Désigne tout **traitement** muni d'un **nom** (instruction Gibiane) qui construit un ou plusieurs **objets nouveaux** à partir d'un ou plusieurs objets existants

- Noms des opérateurs

Imposés à l'utilisateur

Ce sont des instructions Gibiane

Insensibles à la casse

Cast3M ne lit que les **4 premiers caractères** :

DROITE $\Leftrightarrow$ **DROI**

Quelques exceptions : forme abrégée

DROI $\Leftrightarrow$ **D**

CERC $\Leftrightarrow$ **C**

poin2 = **POIN** 28. 3. ; $\Leftrightarrow$ poin2 = 28. 3. ;

obj3 = **MOT** 'Hello' ; $\Leftrightarrow$ obj3 = 'Hello' ;

Gibiane : les opérateurs

- Exemples d'appel à un opérateur (invocation)

Cas courants : 1 objet à gauche du =

obj1 = OPER obj2 ;

obj3 = OPER obj4 obj5 ;

obj6 = obj7 OPER obj8 obj9 ;

Cas exceptionnels : plusieurs objets à gauche du =

obj1 obj2 obj3 = OPER obj4 obj5 ;

Gibiane : les opérateurs

- L'ordre des opérandes

Est **indifférent** si les opérandes sont de **type différents**
(sauf exception dans la documentation)

Est **important** si plusieurs opérandes du **même type**

- Surcharge d'un objet

Toujours possible, l'ancien objet disparaît

A = 'Hello' ; A est du type MOT

B = 28 ;

C = 3 ;

A = B**C ; A est du type ENTIER et vaut 21952

- Pièges classiques

Nom d'objet = nom d'opérateur ⇒ appel à l'opérateur impossible !
sauf si on l'appelle entre quotes et en capitales

A = 'OPER' B C ;

Objet nommé C ou D

Gibiane : les directives

- Définition
Commande sans symbole d'affectation =
Ne crée pas de nouvel objet

- Exemples

```
OPTI  'DIME' 3  'ELEM' 'CUB8' ;  
OPTI  'TITR' 'Maillage de la piece' ;  
  
DEPL mail1 'PLUS' (28. -0.3 20.03) ;
```

Gibiane : les procédures

- Définition

Ensemble nommé de **commandes Gibiane** muni d'une liste d'opérandes d'entrée et de sortie

Analogue à une subroutine Fortran ou à une fonction C

- Nom des procédures

Comme un objet ordinaire (une procédure est un objet de type PROCEDUR)

- Déclaration

```
DEBP  MATHS    n1* 'ENTIER'    x1* 'FLOTTANT' ;  
      add2 = n1 + x1 ;  
      dif2 = n1 - x1 ;  
      mul2 = n1 * x1 ;  
FINP  add2  dif2  mul2 ;
```


Gibiane : les procédures

- Invocation

Comme un opérateur ou une directive ordinaire

a b c = **MATHS** 21 11.0 ;

- Il existe des procédures pré-cablées dans Cast3M

PASAPAS calculs non linéaires

FLAMBAGE calculs de flambage

DYNAMIC calculs dynamiques

THERMIC calculs thermiques

G_THETA calcul d'intégrales J et FIC (rupture)

...

Consultez la documentation :

<https://www-cast3m.cea.fr/index.php?page=notices>

Gibiane : les procédures

- Pièges classiques

FINP manquant

⇒ arrêt de Cast3M + message d'erreur parfois difficile à interpréter

FINP présent mais ; manquant

⇒ arrêt de Cast3M + message d'erreur parfois difficile à interpréter

Invocation d'une procédure **avant** qu'elle ne soit **déclarée**

⇒ arrêt de Cast3M + message de l'opérateur =

Gibiane : quelques instructions utiles

- Débogage

INFO OPER ;

⇒ affiche la notice d'un opérateur/directive/procédure

OPTI 'DONN' 5 ;

⇒ arrêt de la lecture du fichier .dgibi, passage en **mode interactif**

OPTI 'DONN' 3 ;

⇒ reprise de la lecture du fichier .dgibi

LIST obj1 ;

⇒ affiche le contenu de l'objet

LIST 'RESU' obj1 ;

⇒ affiche un **résumé** du contenu de l'objet

OPTI 'DEBU' 1 ;

⇒ accès aux variables locales des procédures

TRAC obj1 ;

⇒ affiche l'objet

MESS 'Hello' ;

⇒ écrit un message

Documentation


- Notices des opérateurs/directives/procédures
 1. Utiliser la directive INFO : INFO 'DROI' ;
 2. Consulter la page html locale
située dans le répertoire d'installation
sur Linux : /home/user/CAST3M_2024/doc/index.html
sur Windows : C:\Cast3M\PCW_24\doc\index.html
 3. Consulter le site web
<https://www-cast3m.cea.fr/index.php?page=notices>
attention, il s'agit de la version du jour !
- Manuels utilisateurs
sur le site web, à l'onglet "Documentation"

Organisation d'un calcul élément-finis

1. Choix de la géométrie et du maillage
 - 1.1 Définition des points lignes, surfaces, volumes
 - 1.2 Discrétisation
2. Définition du modèle mathématique
 - 2.1 Modèle (type d'analyse, formulation, comportement matériau, type d'élément)
 - 2.2 Propriétés matérielles (module d'Young, masse volumique, ...)
 - 2.3 Propriétés géométriques (épaisseur, moments quadratiques, ...)
 - 2.4 Conditions aux limites/chargements
 - 2.5 Conditions initiales
3. Résolution du problème discrétisé
 - 3.1 Calcul des matrices de rigidité/masse élémentaires
 - 3.2 Assemblage des matrices
 - 3.3 Application des conditions limites/chargements
 - 3.4 Résolution du système d'équations
4. Analyse et post-traitement des résultats
 - 4.1 Calcul de quantités locales (déplacement, contraintes, déformation, ...)
 - 4.2 Calcul de quantités globales (déformation maximale, charge limite, ...)

Géométrie

- Dimensions de la pièce



1.1 Maillage non structuré

- Objectif : créer un **maillage paramétré** de la structure
 - ⇒ maillage non structuré
 - ⇒ paramètre : taille de maille globale
 - ⇒ éléments triangles/tétraèdres
- Méthode :
 1. placer des points guides
 2. mailler le contour fermé
 3. mailler la surface par remplissage
 4. mailler le volume par extrusion

1.1 Maillage non structuré

- Options générales et paramètres

```
** Options generales / general options
OPTI 'DIME' 3 ;

** Parametres geometriques / geometrical parameters
l = 0.3 ;
h = 0.1 ;
r = 0.035 ;
e = 0.02 ;

** Taille de maille globale / Global mesh element size
OPTI 'DENS' (e / 2.) ;
```

Nouveaux objets : ENTIER, FLOTTANT, MOT

1.1 Maillage non structuré

- Création de points

```
** Quelques points / Some points
p1  = 0. 0. 0. ;
p2  = 1  0. 0. ;
p3  = (1 + (h / 2.)) 0. (h / 2.) ;
p4  = 1  0. h ;
p5  = 0. 0. h ;
p6  = 1  0. (h / 2.) ;
p7  = p6 PLUS (r  0. 0.) ;
p8  = p6 PLUS (0. 0. r ) ;
p9  = p6 MOIN (r  0. 0.) ;
p10 = p6 MOIN (0. 0. r ) ;
```

Nouvel objet : POINT

1.1 Maillage non structuré

- Maillage des lignes et de contours fermés


```
** Option generale pour mailler en elements TE10 /  
* General option for meshing with TE10 elements  
OPTI 'ELEM' 'TE10' ;  
  
** Maillage du contour / Contour mesh  
112 = DROI p1 p2 ;  
c23 = CERC p2 p6 p3 ;  
c34 = CERC p3 p6 p4 ;  
145 = DROI p4 p5 ;  
151 = DROI p5 p1 ;  
cex = 112 ET c23 ET c34 ET 145 ET 151 ;
```

Nouvel objet : MAILLAGE

1.1 Maillage non structuré

- Maillage des lignes et de contours fermés

```
** Option generale pour mailler en elements TE10 /  
* General option for meshing with TE10 elements  
OPTI 'ELEM' 'TE10' ;  
  
** Maillage du contour / Contour mesh  
112 = DROI p1 p2 ;  
c23 = CERC p2 p6 p3 ;  
c34 = CERC p3 p6 p4 ;  
145 = DROI p4 p5 ;  
151 = DROI p5 p1 ;  
cex = 112 ET c23 ET c34 ET 145 ET 151 ;  
  
TRAC cex ;
```



Nouvel objet : MAILLAGE


1.1 Maillage non structuré

- Maillage des lignes et de contours fermés

```
** Option generale pour mailler en elements TE10 /  
* General option for meshing with TE10 elements  
OPTI 'ELEM' 'TE10' ;
```

```
** Maillage du contour / Contour mesh  
l12 = DROI p1 p2 ;  
c23 = CERC p2 p6 p3 ;  
c34 = CERC p3 p6 p4 ;  
l45 = DROI p4 p5 ;  
l51 = DROI p5 p1 ;  
cex = l12 ET c23 ET c34 ET l45 ET l51 ;  
  
TRAC cex ;
```

👉 *Maillage du cercle de centre p6*



Nouvel objet : MAILLAGE

1.1 Maillage non structuré

- Maillage des lignes et de contours fermés


```
** Option generale pour mailler en elements TE10 /  
* General option for meshing with TE10 elements  
OPTI 'ELEM' 'TE10' ;
```

```
** Maillage du contour / Contour mesh  
l12 = DROI p1 p2 ;  
c23 = CERC p2 p6 p3 ;  
c34 = CERC p3 p6 p4 ;  
l45 = DROI p4 p5 ;  
l51 = DROI p5 p1 ;  
cex = l12 ET c23 ET c34 ET l45 ET l51 ;  
  
TRAC cex ;
```

👉 *Maillage du cercle de centre p6*

```
p11 = p6 PLUS (0. -1. 0.) ;  
cin = CERC 'ROTA' 360. p7 p6 p11 'ELIM' ;  
  
TRAC (cex ET cin) ;
```

Nouvel objet : MAILLAGE



1.1 Maillage non structuré


- Maillage de la surface (maillage libre depuis le contour fermé)

```
** Maillage surfacique / Surface mesh  
sur1 = SURF (cex ET cin) 'PLAN' ;
```

1.1 Maillage non structuré

- Maillage de la surface (maillage libre depuis le contour fermé)


```
** Maillage surfacique / Surface mesh  
sur1 = SURF (cex ET cin) 'PLAN' ;  
  
TRAC sur1 ;
```



1.1 Maillage non structuré

- Maillage de la surface (maillage libre depuis le contour fermé)


```
** Maillage surfacique / Surface mesh  
sur1 = SURF (cex ET cin) 'PLAN' ;  
  
TRAC sur1 ;  
  
cin = INVE cin ;  
sur1 = SURF (cex ET cin) 'PLAN' ;
```



1.1 Maillage non structuré

- Maillage de la surface (maillage libre depuis le contour fermé)


```
** Maillage surfacique / Surface mesh  
sur1 = SURF (cex ET cin) 'PLAN' ;  
  
TRAC sur1 ;  
  
cin = INVE cin ;  
sur1 = SURF (cex ET cin) 'PLAN' ;  
  
TRAC sur1 ;
```



1.1 Maillage non structuré

- Maillage de la surface (maillage libre depuis le contour fermé)

```
** Maillage surfacique / Surface mesh  
sur1 = SURF (cex ET cin) 'PLAN' ;  
  
TRAC sur1 ;  
  
cin = INVE cin ;  
sur1 = SURF (cex ET cin) 'PLAN' ;  
  
TRAC sur1 ;
```




- Maillage du volume

👉 Mailler le volume par translation (voir : VOLU)

1.1 Maillage non structuré

- Maillage de la surface (maillage libre depuis le contour fermé)


```
** Maillage surfacique / Surface mesh  
sur1 = SURF (cex ET cin) 'PLAN' ;  
  
TRAC sur1 ;  
  
cin = INVE cin ;  
sur1 = SURF (cex ET cin) 'PLAN' ;  
  
TRAC sur1 ;
```



- Maillage du volume

👉 *Mailler le volume par translation (voir : VOLU)*


```
** Maillage volumique / Volume mesh  
vol1 = sur1 VOLU 'TRAN' (0. e 0.) ;  
  
TRAC vol1 ;
```



1.1 Maillage non structuré

- Maillage de la surface (maillage libre depuis le contour fermé)


```
** Maillage surfacique / Surface mesh  
sur1 = SURF (cex ET cin) 'PLAN' ;  
  
TRAC sur1 ;  
  
cin = INVE cin ;  
sur1 = SURF (cex ET cin) 'PLAN' ;  
  
TRAC sur1 ;
```



- Maillage du volume

👉 *Mailler le volume par translation (voir : VOLU)*


```
** Maillage volumique / Volume mesh  
vol1 = sur1 VOLU 'TRAN' (0. e 0.) ;  
  
TRAC vol1 ;  
  
TRAC 'CACH' vol1 ;
```



1.1 Maillage non structuré

- Maillage de la surface (maillage libre depuis le contour fermé)


```
** Maillage surfacique / Surface mesh  
sur1 = SURF (cex ET cin) 'PLAN' ;  
  
TRAC sur1 ;  
  
cin = INVE cin ;  
sur1 = SURF (cex ET cin) 'PLAN' ;  
  
TRAC sur1 ;
```



- Maillage du volume

👉 *Mailler le volume par translation (voir : VOLU)*

```
** Maillage volumique / Volume mesh  
vol1 = sur1 VOLU 'TRAN' (0. e 0.) ;  
  
TRAC vol1 ;  
  
TRAC 'CACH' vol1 ;
```




- Le maillage est fait de prismes !

1.1 Maillage non structuré

- Modification du type d'éléments

```
** Recuperation de faces / Face recovery  
sur2 = vol1 FACE 2 ;  
sur3 = vol1 FACE 3 ;  
  
TRAC sur3 ;
```



1.1 Maillage non structuré

- Modification du type d'éléments

```
** Recuperation de faces / Face recovery
```

```
sur2 = vol1 FACE 2 ;
```


```
sur3 = vol1 FACE 3 ;
```

```
TRAC sur3 ;
```

```
** Changement du type d'element /
```

```
* Change element type
```

```
sur3 = CHAN 'TRI6' sur3 ;
```



1.1 Maillage non structuré

- Modification du type d'éléments

```
** Recuperation de faces / Face recovery
```

```
sur2 = vol1 FACE 2 ;
```

```
sur3 = vol1 FACE 3 ;
```


```
TRAC sur3 ;
```

```
** Changement du type d'element /
```

```
* Change element type
```

```
sur3 = CHAN 'TRI6' sur3 ;
```

```
TRAC sur3 ;
```



1.1 Maillage non structuré

- Modification du type d'éléments

```
** Recuperation de faces / Face recovery
```

```
sur2 = vol1 FACE 2 ;
```

```
sur3 = vol1 FACE 3 ;
```

```
TRAC sur3 ;
```


```
** Changement du type d'element /
```

```
* Change element type
```

```
sur3 = CHAN 'TRI6' sur3 ;
```

```
TRAC sur3 ;
```

🔍 Mailler le volume dans la surface enveloppe (voir : VOLU)



1.1 Maillage non structuré

- Modification du type d'éléments

```
** Recuperation de faces / Face recovery
```

```
sur2 = vol1 FACE 2 ;
```

```
sur3 = vol1 FACE 3 ;
```

```
TRAC sur3 ;
```

```
** Changement du type d'element /
```

```
* Change element type
```

```
sur3 = CHAN 'TRI6' sur3 ;
```

```
TRAC sur3 ;
```


🔧 Mailler le volume dans la surface enveloppe (voir : VOLU)

```
** Maillage volumique / Volume mesh
```

```
env1 = sur1 ET sur2 ET sur3 ;
```

```
vol1 = VOLU env1 ;
```

```
TRAC 'CACH' vol1 ;
```



1.2 Maillage structuré

- Objectif : créer un **maillage paramétré** de la structure
 - ⇒ maillage structuré
 - ⇒ paramètre : nombre d'éléments sur les arêtes
 - ⇒ éléments quadrangles/hexaèdres
- Méthode :
 1. placer des points guides
 2. mailler des lignes opposées
 3. mailler des surfaces réglées
 4. mailler le volume par extrusion

1.2 Maillage structuré

- Options et paramètres

```
** Option generale pour mailler en elements CU20 /  
* General option for meshing with CU20 elements  
OPTI 'ELEM' 'CU20' ;  
  
** Nombre d'elements / number of elements  
n15    = 10 ;  
n12    = 20 ;
```


1.2 Maillage structuré

- Options et paramètres

```
** Option generale pour mailler en elements CU20 /  
* General option for meshing with CU20 elements  
OPTI 'ELEM' 'CU20' ;  
  
** Nombre d'elements / number of elements  
n15 = 10 ;  
n12 = 20 ;
```

- Maillage surfacique par translation

```
** Maillage du rectangle /  
* Meshing the rectangle  
l15 = DROI n15 p1 p5 ;  
x13 = 1 - (h / 2.) ;  
sr1 = l15 TRAN n12 (x13 0. 0.) ;  
  
TRAC sr1 ;
```



1.2 Maillage structuré

- Récupération de sous zones

```
** Recuperation de lignes et points /  
* Edges and points recovery  
11213 = sr1 COTE 3 ;  
p12   = 11213 POIN 'INITIAL' ;  
p13   = 11213 POIN 'PROCHE' (x13 0. 0.) ;
```

1.2 Maillage structuré

- Récupération de sous zones

```
** Recuperation de lignes et points /  
* Edges and points recovery  
11213 = sr1 COTE 3 ;  
p12   = 11213 POIN 'INITIAL' ;  
p13   = 11213 POIN 'PROCHE' (x13 0. 0.) ;
```

- Maillage de la surface autour du trou

```
** Contour extérieur du trou /  
* External contour of the hole  
11312 = DROI (n15 / 2) p13 p2 CERC n15 p6 p3  
       CERC n15 p6 p4 DROI (n15 / 2) p12 ;  
lig1  = 11213 ET 11312 ;
```

1.2 Maillage structuré

- Récupération de sous zones

```
** Recuperation de lignes et points /  
* Edges and points recovery  
l1213 = sr1 COTE 3 ;  
p12   = l1213 POIN 'INITIAL' ;  
p13   = l1213 POIN 'PROCHE' (x13 0. 0.) ;
```

- Maillage de la surface autour du trou

```
** Contour extérieur du trou /  
* External contour of the hole  
l1312 = DROI (n15 / 2) p13 p2 CERC n15 p6 p3  
       CERC n15 p6 p4 DROI (n15 / 2) p12 ;  
lig1  = l1213 ET l1312 ;
```

👉 Mailler le cercle du trou par PROJection de *lig1*

1.2 Maillage structuré

- Récupération de sous zones


```
** Recuperation de lignes et points /  
* Edges and points recovery  
11213 = sr1 COTE 3 ;  
p12   = 11213 POIN 'INITIAL' ;  
p13   = 11213 POIN 'PROCHE' (x13 0. 0.) ;
```

- Maillage de la surface autour du trou

```
** Contour extérieur du trou /  
* External contour of the hole  
11312 = DROI (n15 / 2) p13 p2 CERC n15 p6 p3  
       CERC n15 p6 p4 DROI (n15 / 2) p12 ;  
lig1  = 11213 ET 11312 ;
```

👉 Mailler le cercle du trou par PROJection de lig1

```
** Projection sur le cercle intérieur /  
* Projection on the inner circle  
cin   = lig1 PROJ 'CONI' p6 'SPHE' p6 p7 ;  
TRAC (lig1 ET cin) ;
```



1.2 Maillage structuré


- Maillage de la surface réglée
 - ☞ *Mailler la surface réglée entre cin et lig1 avec 3 éléments (voir : REGL)*

1.2 Maillage structuré

- Maillage de la surface réglée

☞ *Mailler la surface réglée entre cin et lig1
avec 3 éléments (voir : REGL)*

```
** Surface reglee du trou /  
*   Ruled surface of the hole  
sh1   = cin REGL 3 lig1 ;  
** Maillage surfacique / Surface mesh  
sur1  = sr1 ET sh1 ;
```





1.2 Maillage structuré

- Maillage de la surface réglée

☞ *Mailler la surface réglée entre cin et lig1
avec 3 éléments (voir : REGL)*

```
** Surface reglee du trou /  
* Ruled surface of the hole  
sh1   = cin REGL 3 lig1 ;  
** Maillage surfacique / Surface mesh  
sur1  = sr1 ET sh1 ;  
  
TRAC sur1 ;
```



1.2 Maillage structuré




- Maillage de la surface réglée

👉 *Mailler la surface réglée entre cin et lig1
avec 3 éléments (voir : REGL)*

```
** Surface reglee du trou /  
* Ruled surface of the hole  
sh1   = cin REGL 3 lig1 ;  
** Maillage surfacique / Surface mesh  
sur1  = sr1 ET sh1 ;  
  
TRAC sur1 ;
```

- Maillage du volume par translation

```
** Maillage volumique / Volume mesh  
vol1  = sur1 VOLU 2 'TRAN' (0. e 0.) ;  
  
TRAC 'CACH' vol1 ;
```



1.2 Maillage structuré

- Récupération de sous zones

```
** Recuperation de faces / Faces recovery
sur2  = vol1 FACE 2 ;
sur3  = vol1 FACE 3 ;
* Surface enveloppe
env1  = ENVE vol1 ;
* Aretes / Edges
are1  = ARET vol1 ;
```

1.2 Maillage structuré

- Récupération de sous zones

```
** Recuperation de faces / Faces recovery
sur2  = vol1 FACE 2 ;
sur3  = vol1 FACE 3 ;
* Surface enveloppe
env1  = ENVE vol1 ;
* Aretes / Edges
are1  = ARET vol1 ;
```

- Sauvegarde des données

```
OPTI 'SAUV' 'formation_debutant_1_maillage.sauv' ;
SAUV ;

FIN ;
```

Thermique

Rappels

- Équation de la chaleur

Forme locale

$$\rho c_p \frac{\partial T}{\partial t} + \operatorname{div} \underbrace{(-\lambda \operatorname{grad}(T))}_{\underline{\phi}} = q_{\text{imp}} \quad \text{sur } V$$

- Conditions aux limites

Températures imposées $T = T_{\text{imp}}$

Flux imposés

$$\underline{\phi} \cdot \underline{n} = \phi_{\text{imp}} + \underbrace{h(T_f - T)}_{\text{convection}} + \underbrace{\varepsilon \sigma (T_{\infty}^4 - T^4)}_{\text{rayonnement}} \quad \begin{array}{l} \text{sur } \partial V_T \\ \text{sur } \partial V_{\phi} \end{array}$$

- Avec

T température

$\underline{\phi}$ densité de flux de chaleur

q_{imp} source de chaleur volumique

t temps

λ conductivité thermique

ρ masse volumique

c_p capacité thermique massique

Thermique

Rappels

- Éléments finis :

$$T(x) = [N(x)]\{T\} \quad \underline{\text{grad}}(T) = [B(x)]\{T\}$$

Formulation faible et discrétisée :

$$[C]\{\dot{T}\} + [K]\{T\} = \{P\}$$

avec les matrices :

$$[C] = \int_V \rho c_p [N]^T [N] dV$$

matrice de capacité

$$[K] = \int_V [B]^T [\lambda] [B] dV + \int_{\partial V_\phi} h [N]^T [N] dS$$


matrice de conductivité

et le vecteur chargement nodal équivalent :

$$\{P\} = \int_V [N]^T q_{\text{imp}} dV + \int_{\partial V_\phi} [N]^T (\phi_{\text{imp}} + hT_f + \varepsilon\sigma(T_\infty^4 - T^4)) dS$$

2.1 Problème étudié

- Conduction, régime stationnaire




$$\lambda = 50 \text{ W.m}^{-1}.\text{K}^{-1}$$

$$c_p = 420 \text{ J.K}^{-1}.\text{kg}^{-1}$$


$$\rho = 7800 \text{ kg.m}^{-3}$$

$$T_{t=0} = 0 \text{ }^{\circ}\text{C}$$

- Température imposée



- Flux de chaleur imposé



2.1 Thermique linéaire

Stationnaire, conduction

- Objectif : calcul thermique stationnaire
en températures et flux imposés

$$[C]\{\dot{T}\} + [K]\{T\} = \{P\} \quad \Rightarrow \quad \text{Système linéaire}$$

- Méthode :
 - calcul de la matrice de conductivité
 - calcul des chargement nodaux équivalents
 - résolution avec **RESO**

$$\begin{aligned} & [K] \\ & \{P\} \\ & \{T\} = [K]^{-1}\{P\} \end{aligned}$$

2.1 Thermique linéaire

Stationnaire, conduction

- Restitution des objets (maillage, paramètres, ...)

```
OPTI 'REST' 'formation_debutant_1_maillage.sauv' ;  
REST ;
```

👉 *Tous les objets sauvegardés sont chargés en mémoire
Ils sont alors accessibles de suite*

- Nouveaux paramètres

```
** Parametres materiau / material parameters  
lambda = 50. ;  
cp      = 420. ;  
rho     = 7800. ;  
  
** Parametres des conditions limites et chargement /  
* Parameters for boundary conditions and loading  
t0      = 0. ;  
timp    = 250. ;  
fimp    = -40000. ;
```

2.1 Thermique linéaire

Stationnaire, conduction

- Formulation mathématique (conduction)

```
** Modele de conduction / Conduction model
mot = MODE vol1 'THERMIQUE' 'CONDUCTION' ;
mat = MATE mot 'K' lambda 'C' cp 'RHO' rho 'TINI' t0 ;
```

2.1 Thermique linéaire

Stationnaire, conduction

- Formulation mathématique (conduction)

```
** Modele de conduction / Conduction model
mot = MODE vol1 'THERMIQUE' 'CONDUCTION' ;
mat = MATE mot 'K' lambda 'C' cp 'RHO' rho 'TINI' t0 ;
```

- Matrice de conductivité

```
** Matrice de conductivite / Conductivity matrix
con = COND mot mat ;
```

$$[K] = \int_V [B]^T [\lambda] [B] dV + \int_{\partial V_\phi} h [N]^T [N] dS$$


Nouveaux objets : MMODEL, MCHAML, RIGIDITE

2.1 Thermique linéaire

Stationnaire, conduction

- Conditions aux limites

```
** Recuperation de surfaces /  
* Recovery of some surfaces  
pt1  = env1 POIN 'PLAN' p1 p5 (0. 1. 0.) ;  
sut1 = env1 ELEM 'APPU' 'STRI' pt1 ;  
pint = env1 POIN 'CYLI' p6 p11 p7 ;  
sint = env1 ELEM 'APPU' 'STRI' pint ;  
  
TRAC 'FACE' (sint COUL 'BLEU') are1 ;
```




2.1 Thermique linéaire

Stationnaire, conduction

- Conditions aux limites

```
** Recuperation de surfaces /  
* Recovery of some surfaces  
pt1  = env1 POIN 'PLAN' p1 p5 (0. 1. 0.) ;  
sut1 = env1 ELEM 'APPU' 'STRI' pt1 ;  
pint = env1 POIN 'CYLI' p6 p11 p7 ;  
sint = env1 ELEM 'APPU' 'STRI' pint ;  
  
TRAC 'FACE' (sint COUL 'BLEU') are1 ;  
  
** Matrice de blocage / Boundary condition matrix  
blt = BLOQ 'T' sint ;  
  
** Vecteur associe au blocage /  
* Vector associated to the bc matrix  
pti = DEPI blt timp ;
```



2.1 Thermique linéaire

Stationnaire, conduction

- Conditions aux limites

```

** Recuperation de surfaces /
* Recovery of some surfaces
pt1  = env1 POIN 'PLAN' p1 p5 (0. 1. 0.) ;
sut1 = env1 ELEM 'APPU' 'STRI' pt1 ;
pint = env1 POIN 'CYLI' p6 p11 p7 ;
sint = env1 ELEM 'APPU' 'STRI' pint ;


TRAC 'FACE' (sint COUL 'BLEU') are1 ;

** Matrice de blocage / Boundary condition matrix
blt = BLOQ 'T' sint ;

** Vecteur associe au blocage /
* Vector associated to the bc matrix
pti = DEPI blt timp ;

** Vecteur flux impose a gauche /
* Flow vector imposed on the left
pfi = FLUX mot sut1 fimp ;

```



$$\{P\} = \int_V [N]^T q_{\text{imp}} dV + \int_{\partial V_\phi} [N]^T (\phi_{\text{imp}} + hT_f + \varepsilon\sigma(T_\infty^4 - T^4)) dS$$

2.1 Thermique linéaire

Stationnaire, conduction

- Résolution et affichage des résultats


```
** Resolution / Solving  
t21 = RESO (con ET blt) (pti ET pfi) ;
```

2.1 Thermique linéaire

Stationnaire, conduction

- Résolution et affichage des résultats

```
** Resolution / Solving  
t21 = RESO (con ET blt) (pti ET pfi) ;  
  
TRAC t21 vol1 ;
```




2.1 Thermique linéaire

Stationnaire, conduction

- Résolution et affichage des résultats

```
** Resolution / Solving  
t21 = RESO (con ET blt) (pti ET pfi) ;  
  
TRAC t21 vol1 ;  
  
TRAC t21 vol1 are1 ;
```



2.1 Thermique linéaire

Stationnaire, conduction

- Résolution et affichage des résultats


```
** Resolution / Solving
t21 = RESO (con ET blt) (pti ET pfi) ;

TRAC t21 vol1 ;

TRAC t21 vol1 are1 ;

liso1 = PROG 0. 'PAS' (timp / 15.) timp ;

TRAC t21 vol1 are1 liso1 ;
```




T
< 2.50E+02
> 3.20E+01

2.50E+02
2.33E+02
2.17E+02
2.00E+02
1.83E+02
1.67E+02
1.50E+02
1.33E+02
1.17E+02
1.00E+02
83.
67.
50.
33.
17.
0.0


Nouvel objet : LISTREEL

2.2 Problème étudié


- Conduction, régime stationnaire
- Température imposée




- Flux de chaleur imposé



- Convection



- Source volumique



2.2 Thermique linéaire

Stationnaire, conduction, convection, source

- Nouveaux paramètres

```
simp      = 2600.E3 ;  
hconv     = 240. ;  
tconv     = -80. ;
```

2.2 Thermique linéaire


Stationnaire, conduction, convection, source

- Nouveaux paramètres

```
simp      = 2600.E3 ;  
hconv     = 240. ;  
tconv     = -80. ;
```

- Récupération de sous zones

```
** Recuperation de surfaces /  
* Recovery of some surfaces  
z          = COOR 3 env1 ;  
ptcon = z POIN 'EGAL' 0. ;  
sucon = env1 ELEM 'APPU' 'STRI' ptcon ;  
TRAC 'FACE' (sucon COUL 'TURQ') are1 ;
```



2.2 Thermique linéaire



Stationnaire, conduction, convection, source

- Nouveaux paramètres

```
simp    = 2600.E3 ;  
hconv   = 240. ;  
tconv   = -80. ;
```

- Récupération de sous zones

```
** Recuperation de surfaces /  
* Recovery of some surfaces  
z      = COOR 3 env1 ;  
ptcon  = z POIN 'EGAL' 0. ;  
sucon  = env1 ELEM 'APPU' 'STRI' ptcon ;  
  
TRAC 'FACE' (sucon COUL 'TURQ') are1 ;  
  
** Recuperation de volumes /  
* Recovery of some volumes  
x      = COOR 1 vol1 ;  
ptso   = x POIN 'COMPRIS' (0.33 * 1) (0.51 * 1) ;  
elso   = vol1 ELEM 'APPU' 'STRI' ptso ;  
  
TRAC 'FACE' (elso COUL 'VERT') are1 ;
```



2.2 Thermique linéaire

Stationnaire, conduction, convection, source

- Formulation mathématique (convection)

```
** Modele de convection / Convection model  
moc = MODE sucon 'THERMIQUE' 'CONVECTION' ;  
mac = MATE moc 'H' hconv ;
```

2.2 Thermique linéaire

Stationnaire, conduction, convection, source

- Formulation mathématique (convection)

```
** Modele de convection / Convection model
moc = MODE sucon 'THERMIQUE' 'CONVECTION' ;
mac = MATE moc 'H' hconv ;
```

- Matrice de conductivité (mais pour la convection !)

```
** Matrice de conductivite pour la convection /
* Conductivity matrix for convection
conh = COND moc mac ;
```

$$[K] = \int_V [B]^T [\lambda] [B] dV + \int_{\partial V_\phi} h [N]^T [N] dS$$

2.2 Thermique linéaire

Stationnaire, conduction, convection, source

- Formulation mathématique (convection)

```
** Modele de convection / Convection model
moc = MODE sucon 'THERMIQUE' 'CONVECTION' ;
mac = MATE moc 'H' hconv ;
```

- Matrice de conductivité (mais pour la convection !)

```
** Matrice de conductivite pour la convection /
* Conductivity matrix for convection
conh = COND moc mac ;
```

$$[K] = \int_V [B]^T [\lambda] [B] dV + \int_{\partial V_\phi} h [N]^T [N] dS$$

- Chargement nodal équivalent (convection)

```
** Vecteur flux pour la convection /
* Flow vector for convection
chtc = MANU 'CHPO' sucon 'T' tconv ;
ph = CONV moc mac chtc ;
```

$$\{P\} = \int_V [N]^T q_{\text{imp}} dV + \int_{\partial V_\phi} [N]^T (\phi_{\text{imp}} + h T_f + \varepsilon \sigma (T_\infty^4 - T^4)) dS$$

2.2 Thermique linéaire

Stationnaire, conduction, convection, source

- Chargement nodal équivalent (source)

```
** Source de chaleur
```

```
psi = SOUR (REDU mot elso) simp elso ;
```

$$\{P\} = \int_V [N]^T q_{\text{imp}} dV + \int_{\partial V_\phi} [N]^T (\phi_{\text{imp}} + h T_f + \varepsilon \sigma (T_\infty^4 - T^4)) dS$$

2.2 Thermique linéaire

Stationnaire, conduction, convection, source

- Chargement nodal équivalent (source)

```
** Source de chaleur
```

```
psi = SOUR (REDU mot else) simp else ;
```


$$\{P\} = \int_V [N]^T q_{\text{imp}} dV + \int_{\partial V_\phi} [N]^T (\phi_{\text{imp}} + h T_f + \varepsilon \sigma (T_\infty^4 - T^4)) dS$$

- Résolution et affichage des résultats

```
** Resolution / Solving
```

```
t22 = RESO (con ET conh ET blt) (pti ET pfi ET psi ET ph) ;
```

```
TRAC t22 vol1 are1 ;
```



Remarques : champs par éléments (MCHAML)

- Objet MCHAML


Représente un champ de valeurs exprimées **dans les ÉLÉMENTS** d'un maillage

Exemples :

- paramètres matériau
- gradient de températures, déplacements...
- contraintes, déformations
- variables internes
- et bien d'autres...

Quelques caractéristiques :

- plusieurs points support possibles :
 - points d'intégration des contraintes
 - point d'intégration de la rigidité
 - points d'intégration de la masse
 - centre de gravité
 - nœuds
- interpolé par les fonctions $[N(x)]$
- non continu d'un élément à l'autre



Remarques : champs par points (CHPOINT)

- Objet CHPOINT



Représente un champ de valeurs exprimées aux **POINTS (nœuds)**

Exemples :

- champ scalaire de température
- champ vectoriel de déplacement (3 composantes)
- champ vectoriel de coordonnées des nœuds
- chargement nodal équivalent
- et bien d'autres...

Quelques caractéristiques :


- une seule valeur possible par nœud
- ne dépend pas du maillage, seulement des nœuds !
- continu sur le maillage




3 Problème étudié

- Conduction, régime transitoire


- Température imposée




- Flux de chaleur imposé



- Convection



- Source volumique



3 Thermique linéaire

Transitoire, conduction, convection, source

- Objectif : calcul thermique transitoire

$$[C]\{\dot{T}\} + [K]\{T\} = \{P\}$$

- Méthode :
 - description temporelle des chargements
 - conditions initiales
 - résolution avec la procédure PASAPAS

3 Thermique linéaire

Transitoire, conduction, convection, source

- Intervalle de temps

```
** Instant final / Final time  
tfinal = 15000. ;
```

3 Thermique linéaire

Transitoire, conduction, convection, source

- Intervalle de temps

```
** Instant final / Final time  
tfinal = 15000. ;
```

- Chargements : description temporelle

```
** Descriptions temporelles des chargements /  
* Time descriptions of the loads  
* 1) Temperature imposee / Imposed temperature  
list1 = PROG 0. (0.1 * tfinal) tfinal ;  
list2 = PROG (t0 / timp) 1. 1. ;  
evt1 = EVOL 'MANU' list1 list2 ;  
chatimp = CHAR 'TIMP' pti evt1 ;
```

Nouveaux objets : EVOLUTIO_n, CHARGEMEnt

3 Thermique linéaire

Transitoire, conduction, convection, source

- Intervalle de temps

```
** Instant final / Final time  
tfinal = 15000. ;
```

- Chargements : description temporelle

```
** Descriptions temporelles des chargements /  
* Time descriptions of the loads  
* 1) Temperature imposee / Imposed temperature  
list1 = PROG 0. (0.1 * tfinal) tfinal ;  
list2 = PROG (t0 / timp) 1. 1. ;  
evt1 = EVOL 'MANU' list1 list2 ;  
chatimp = CHAR 'TIMP' pti evt1 ;  
  
* 2) Flux surfacique impose / Imposed surface flow  
list3 = PROG 0. (0.3 * tfinal) (0.5 * tfinal) tfinal ;  
list4 = PROG 0. 0. 1. 1. ;  
evt2 = EVOL 'MANU' list3 list4 ;  
chafimp = CHAR 'Q' pfi evt2 ;
```

Nouveaux objets : EVOLUTiOn, CHARGEMEnt

3 Thermique linéaire

Transitoire, conduction, convection, source

- Intervalle de temps

```
** Instant final / Final time  
tfinal = 15000. ;
```

- Chargements : description temporelle

```
** Descriptions temporelles des chargements /  
* Time descriptions of the loads  
* 1) Temperature imposee / Imposed temperature  
list1 = PROG 0. (0.1 * tfinal) tfinal ;  
list2 = PROG (t0 / timp) 1. 1. ;  
evt1 = EVOL 'MANU' list1 list2 ;  
chatimp = CHAR 'TIMP' pti evt1 ;  
  
* 2) Flux surfacique impose / Imposed surface flow  
list3 = PROG 0. (0.3 * tfinal) (0.5 * tfinal) tfinal ;  
list4 = PROG 0. 0. 1. 1. ;  
evt2 = EVOL 'MANU' list3 list4 ;  
chafimp = CHAR 'Q' pfi evt2 ;  
  
* 3) Source volumique / Volume source  
chasour = CHAR 'Q' psi evt2 ;  
  
* 4) Convection  
chaconv = CHAR 'TECO' chtc ;
```

Nouveaux objets : EVOLUTiOn, CHARGEMEnt

3 Thermique linéaire

Transitoire, conduction, convection, source

- Construction d'une table pour PASAPAS

```
** Resolution avec la procedure PASAPAS /  
* Solving with the PASAPAS procedure  
delta_t                = tfinal / 100. ;  
tab1                    = TABL ;  
tab1 . 'MODELE'         = mot ET moc ;  
tab1 . 'CARACTERISTIQUES' = mat ET mac ;  
tab1 . 'BLOCAGES_THERMIQUES' = blt ;  
tab1 . 'CHARGEMENT'     = chatimp ET chafimp ET  
                        chasour ET chaconv ;  
tab1 . 'TEMPS_CALCULES' = PROG 0. 'PAS' delta_t tfinal ;
```

Nouvel objet : TABLE

3 Thermique linéaire

Transitoire, conduction, convection, source

- Construction d'une table pour PASAPAS

```
** Resolution avec la procedure PASAPAS /  
* Solving with the PASAPAS procedure  
delta_t                = tfinal / 100. ;  
tab1                   = TABL ;  
tab1 . 'MODELE'        = mot ET moc ;  
tab1 . 'CARACTERISTIQUES' = mat ET mac ;  
tab1 . 'BLOCAGES_THERMIQUES' = blt ;  
tab1 . 'CHARGEMENT'     = chatimp ET chafimp ET  
                        chasour ET chaconv ;  
tab1 . 'TEMPS_CALCULES' = PROG 0. 'PAS' delta_t tfinal ;
```

- Résolution avec PASAPAS

```
PASAPAS tab1 ;
```

Nouvel objet : TABLE

3 Thermique linéaire

Transitoire, conduction, convection, source

- Post traitement : évolutions temporelles

```
** Post traitement / Post processing
* 1) Evolution temporelle temperature en 4 noeuds /
*   Time evolution of temperature on 4 nodes
pa = vol1 POIN 'PROC' (0.                (e / 2.) 0.) ;
pb = vol1 POIN 'PROC' ((0.45 * 1)        (e / 2.) h ) ;
pc = vol1 POIN 'PROC' ((0.8 * 1)         (e / 2.) (h / 2.)) ;
pd = vol1 POIN 'PROC' ((1 + (h / 2.)) (e / 2.) (h / 2.)) ;
```


3 Thermique linéaire

Transitoire, conduction, convection, source

- Post traitement : évolutions temporelles

```
** Post traitement / Post processing
* 1) Evolution temporelle temperature en 4 noeuds /
*   Time evolution of temperature on 4 nodes
pa = vol1 POIN 'PROC' (0. (e / 2.) 0.) ;
pb = vol1 POIN 'PROC' ((0.45 * 1) (e / 2.) h) ;
pc = vol1 POIN 'PROC' ((0.8 * 1) (e / 2.) (h / 2.)) ;
pd = vol1 POIN 'PROC' ((1 + (h / 2.)) (e / 2.) (h / 2.)) ;

TRAC (are1 ET (pa ET pb ET pc ET pd)) ;
```



3 Thermique linéaire

Transitoire, conduction, convection, source

- Post traitement : évolutions temporelles

```



** Post traitement / Post processing
* 1) Evolution temporelle temperature en 4 noeuds /
*   Time evolution of temperature on 4 nodes
pa = vol1 POIN 'PROC' (0. (e / 2.) 0.) ;
pb = vol1 POIN 'PROC' ((0.45 * 1) (e / 2.) h) ;
pc = vol1 POIN 'PROC' ((0.8 * 1) (e / 2.) (h / 2.)) ;
pd = vol1 POIN 'PROC' ((1 + (h / 2.)) (e / 2.) (h / 2.)) ;

TRAC (are1 ET (pa ET pb ET pc ET pd)) ;

eva = EVOL 'BLEU' 'TEMP' tab1 'TEMPERATURES' 'T' pa ;
evb = EVOL 'VERT' 'TEMP' tab1 'TEMPERATURES' 'T' pb ;
evc = EVOL 'ORAN' 'TEMP' tab1 'TEMPERATURES' 'T' pc ;
evd = EVOL 'ROUG' 'TEMP' tab1 'TEMPERATURES' 'T' pd ;

DESS (eva ET evb ET evc ET evd) ;

```




3 Thermique linéaire

Transitoire, conduction, convection, source

- Post traitement : boucle de tracés

```
* 2) Boucle : trace des temperature a chaque instant  
* Loop: plotting temperature for each time step
```

```
n1 = DIME (tab1 . 'TEMPERATURES' ) ;  
REPE b1 n1 ;  
  ti   = tab1 . 'TEMPERATURES' . (&b1 - 1) ;  
  tpsi = tab1 . 'TEMPS' . (&b1 - 1) ;  
  prc  = ENTI (100. * tpsi / tfinal) ;  
  titi = CHAI '[3] Temperature au temps' ' ' tpsi ' (' prc ' %)' ;  
  
  TRAC ti vol1 are1 liso1 'TITR' titi ;  
  
FIN b1 ;
```



[3] Temperature au temps 1.50000E+04 (100 %)

4 Thermique linéaire

Transitoire, conduction, convection, source

- Post traitement : vecteurs flux de chaleur $\underline{\phi} = -\lambda \underline{\text{grad}}(T)$

4 Thermique linéaire

Transitoire, conduction, convection, source

- Post traitement : vecteurs flux de chaleur $\underline{\phi} = -\lambda \underline{\text{grad}}(T)$
- Définition d'une **procédure**

```
DEBP VECFLU  chpo1*'CHPOINT'  model1*'MMODEL'  mater1*'MCHAML' ;
* Gradient de temperature / Gradient of temperature
grad1 = GRAD chpo1 model1 ;
* Changement du type / Type change
grad2 = CHAN 'TYPE' grad1 'CARACTERISTIQUES' ;
* Multiplication par la conductivite / Multiplication by the conductivity
q      = mater1 * grad2 (MOTS 'K ' 'K ' 'K ') (MOTS 'T,X' 'T,Y' 'T,Z')
                                (MOTS 'QX' 'QY' 'QZ') ;

q      = -1. * q ;
* Objet vecteur / Vector object

q      = CHAN 'CHPO' q model1 ;
vec1   = @VECOUL q 2.E-7 (MOTS 'QX' 'QY' 'QZ') ;

FINP vec1 ;
```

Nouveaux objets : PROCEDURE, VECTEUR

4 Thermique linéaire

Transitoire, conduction, convection, source


- Post traitement : vecteurs flux de chaleur $\phi = -\lambda \text{grad}(T)$

**** Utilisation de cette procedure / Using this procedure**

```
tn1 = tab1 . 'TEMPERATURES' . (n1 - 1) ;
```

```
vf1 = VECFLU tn1 mot mat ;
```

```
TRAC vf1 are1 ;
```



1.05E+05
9.35E+04
8.25E+04
7.15E+04
6.05E+04
4.95E+04
3.85E+04
2.75E+04
1.65E+04
5.50E+03

4 Thermique linéaire

Transitoire, conduction, convection, source

- Post traitement : vecteurs flux de chaleur $\phi = -\lambda \text{grad}(T)$

**** Utilisation de cette procedure / Using this procedure**

```
tn1 = tab1 . 'TEMPERATURES' . (n1 - 1) ;
```



```
vf1 = VECFLU tn1 mot mat ;
```

```
TRAC vf1 are1 ;
```

- Tracé de lignes d'iso valeurs

```
OPTI 'ISOV' 'LIGN' ;
```

```
TRAC tn1 vol1 are1 10 ;
```



4 Thermique linéaire

Transitoire, conduction, convection, source

- Post traitement : vecteurs flux de chaleur $\phi = -\lambda \text{grad}(T)$

**** Utilisation de cette procedure / Using this procedure**

```
tn1 = tab1 . 'TEMPERATURES' . (n1 - 1) ;
```

```
vf1 = VECFLU tn1 mot mat ;
```

```
TRAC vf1 are1 ;
```



- Tracé de lignes d'iso valeurs

```
OPTI 'ISOV' 'LIGN' ;
```

```
TRAC tn1 vol1 are1 10 ;
```


- Retour au mode de tracé par défaut

```
OPTI 'ISOV' 'SURF' ;
```



5 Problème étudié

- Conduction, régime transitoire
- Température imposée
- Convection
- Rayonnement
- Flux de chaleur imposé
- Source volumique



5 Thermique non linéaire

Transitoire, conduction, convection, source, rayonnement

- Objectif : calcul thermique transitoire non linéaire

$$[C]\{\dot{T}\} + [K]\{T\} = \{P(T)\}$$

- Méthode :
 - ajout d'un modèle de rayonnement
 - ajout d'un chargement de rayonnement

5 Thermique non linéaire

Transitoire, conduction, convection, source, rayonnement

- Nouveaux paramètres

```
emi      = 0.9 ;  
trayo    = -140. ;
```

- Formulation mathématique (rayonnement)

```
** Modele de rayonnement / Radiation model  
mor = MODE sur2 'THERMIQUE' 'RAYONNEMENT' 'INFINI' ;  
mar = MATE mor 'EMIS' emi ;
```

5 Thermique non linéaire

Transitoire, conduction, convection, source, rayonnement

- Nouveaux paramètres

```
emi      = 0.9 ;  
trayo    = -140. ;
```

- Formulation mathématique (rayonnement)

```
** Modele de rayonnement / Radiation model  
mor = MODE sur2 'THERMIQUE' 'RAYONNEMENT' 'INFINI' ;  
mar = MATE mor 'EMIS' emi ;
```

- Chargement température de rayonnement + description temporelle

```
** Chargement de rayonnement / Radiation load  
chtr    = MANU 'CHPO' sur2 'T' trayo ;  
charay  = CHAR 'TERA' chtr ;
```

5 Thermique non linéaire

Transitoire, conduction, convection, source, rayonnement

- Résolution avec **PASAPAS**

👉 *Remplir la table pour PASAPAS et résoudre*

5 Thermique non linéaire

Transitoire, conduction, convection, source, rayonnement

- Résolution avec **PASAPAS**

👉 *Remplir la table pour PASAPAS et résoudre*

```
** Resolution avec la procedure PASAPAS /  
* Solving with the PASAPAS procedure  
tab1                                = TABL ;  
tab1 . 'MODELE'                    = mot ET moc ET mor ;  
tab1 . 'CARACTERISTIQUES'          = mat ET mac ET mar ;  
tab1 . 'BLOCAGES_THERMIQUES'       = blt ;  
tab1 . 'CHARGEMENT'                = chatimp ET chafimp ET  
                                   chasour ET chaconv ET charay ;  
tab1 . 'TEMPS_CALCULES'             = PROG 0. 'PAS' delta_t tfinal ;  
tab1 . 'CELSIUS'                   = VRAI ;  
PASAPAS tab1 ;
```

5 Thermique non linéaire

Transitoire, conduction, convection, source, rayonnement

- Post traitement : courbes d'évolution temporelles

```
evar = EVOL 'BLEU' 'TEMP' tab1 'TEMPERATURES' 'T' pa ;  
evbr = EVOL 'VERT' 'TEMP' tab1 'TEMPERATURES' 'T' pb ;  
evcr = EVOL 'ORAN' 'TEMP' tab1 'TEMPERATURES' 'T' pc ;  
evdr = EVOL 'ROUG' 'TEMP' tab1 'TEMPERATURES' 'T' pd ;
```


5 Thermique non linéaire

Transitoire, conduction, convection, source, rayonnement

- Post traitement : courbes d'évolution temporelles

```
evar = EVOL 'BLEU' 'TEMP' tab1 'TEMPERATURES' 'T' pa ;  
evbr = EVOL 'VERT' 'TEMP' tab1 'TEMPERATURES' 'T' pb ;  
evcr = EVOL 'ORAN' 'TEMP' tab1 'TEMPERATURES' 'T' pc ;  
evdr = EVOL 'ROUG' 'TEMP' tab1 'TEMPERATURES' 'T' pd ;
```

- Tracé avec légende

```
tleg = TABL ;  
tleg . 1 = 'TIRR' ;  
tleg . 2 = 'TIRR' ;  
tleg . 3 = 'TIRR' ;  
tleg . 4 = 'TIRR' ;  
tleg . 'TITRE' = TABL ;  
tleg . 'TITRE' . 1 = 'PA cond.' ;  
tleg . 'TITRE' . 2 = 'PB cond.' ;  
tleg . 'TITRE' . 3 = 'PC cond.' ;  
tleg . 'TITRE' . 4 = 'PD cond.' ;  
tleg . 'TITRE' . 5 = 'PA cond. + ray.' ;  
tleg . 'TITRE' . 6 = 'PB cond. + ray.' ;  
tleg . 'TITRE' . 7 = 'PC cond. + ray.' ;  
tleg . 'TITRE' . 8 = 'PD cond. + ray.' ;
```

5 Thermique non linéaire

Transitoire, conduction, convection, source, rayonnement

- Post traitement : courbes d'évolution temporelles

```

evar = EVOL 'BLEU' 'TEMP' tab1 'TEMPERATURES' 'T' pa ;
evbr = EVOL 'VERT' 'TEMP' tab1 'TEMPERATURES' 'T' pb ;
evcr = EVOL 'ORAN' 'TEMP' tab1 'TEMPERATURES' 'T' pc ;
evdr = EVOL 'ROUG' 'TEMP' tab1 'TEMPERATURES' 'T' pd ;

```

- Tracé avec légende

```


tleg = TABL ;
tleg . 1 = 'TIRR' ;
tleg . 2 = 'TIRR' ;
tleg . 3 = 'TIRR' ;
tleg . 4 = 'TIRR' ;
tleg . 'TITRE' = TABL ;
tleg . 'TITRE' . 1 = 'PA cond.' ;
tleg . 'TITRE' . 2 = 'PB cond.' ;
tleg . 'TITRE' . 3 = 'PC cond.' ;
tleg . 'TITRE' . 4 = 'PD cond.' ;
tleg . 'TITRE' . 5 = 'PA cond. + ray.' ;
tleg . 'TITRE' . 6 = 'PB cond. + ray.' ;
tleg . 'TITRE' . 7 = 'PC cond. + ray.' ;
tleg . 'TITRE' . 8 = 'PD cond. + ray.' ;

```

```

DESS (eva ET evb ET evc ET evd ET
      evar ET evbr ET evcr ET evdr) 'LEGE' tleg ;

```




5 Thermique non linéaire

Transitoire, conduction, convection, source, rayonnement

- Post traitement : boucle de tracés

```
n1 = DIME (tab1 . 'TEMPERATURES' ) ;  
REPE b1 n1 ;  
  ti    = tab1 . 'TEMPERATURES' . (&b1 - 1) ;  
  tpsi  = tab1 . 'TEMPS' . (&b1 - 1) ;  
  prc   = ENTI (100. * tpsi / tfinal) ;  
  titi  = CHAI '[5] Temperature au temps' ' ' tpsi ' (' prc ' %)' ;  
  
  TRAC ti vol1 are1 liso1 'TITR' titi ;  
  
FIN b1 ;
```



[5] Temperature au temps 1.50000E+04 (100 %)

5 Thermique non linéaire


Transitoire, conduction, convection, source, rayonnement

- Post traitement : boucle de tracés

```
n1 = DIME (tab1 . 'TEMPERATURES' ) ;  
REPE b1 n1 ;  
  ti      = tab1 . 'TEMPERATURES' . (&b1 - 1) ;  
  tpsi    = tab1 . 'TEMPS' . (&b1 - 1) ;  
  prc     = ENTI (100. * tpsi / tfinal) ;  
  titi    = CHAI '[5] Temperature au temps' ' ' tpsi ' (' prc ' %)' ;  
  
  TRAC ti vol1 are1 liso1 'TITR' titi ;  
  
FIN b1 ;
```

- Sauvegarde des données

```
OPTI 'SAUV' 'formation_debutant_2_thermique.sauv' ;  
SAUV ;  
  
FIN ;
```



[5] Temperature au temps 1.50000E+04 (100 %)

Mécanique

Rappels

- Équation d'équilibre statique

Forme locale

$$\operatorname{div}(\underline{\underline{\sigma}}) + \underline{f} = \underline{0} \quad \text{sur } V$$

- Conditions aux limites

Déplacements imposés $\underline{u} = \underline{u}_{\text{imp}} \quad \text{sur } \partial V_u$

Efforts imposés $\underline{\underline{\sigma}} \cdot \underline{n} = \underline{t}_{\text{imp}} \quad \text{sur } \partial V_t$

- Avec

$\underline{u}$ vecteur déplacement

$\underline{\underline{\sigma}}$ tenseur des contraintes

$\underline{f}$ vecteur des forces volumiques

$\underline{n}$ vecteur normal à la surface

Mécanique

Rappels

- Éléments finis : $\underline{u}(x) = [N(x)]\{U\}$

Formulation faible et discrétisée de l'équilibre :

$$\int_V [B]^T \{\sigma\} dV = \{F\}$$

$$\underbrace{\int_V [B]^T \{\sigma\} dV}_{[B]\{\sigma\}} = \underbrace{\int_{\partial V_t} [N]^T \{t\}_{\text{imp}} dS}_{\{F\}_s} + \underbrace{\int_{\partial V_u} [N]^T \{\sigma \cdot n\} dS}_{\{F\}_{\text{reac}}} + \underbrace{\int_V [N]^T \{f\}_s dV}_{\{F\}_v}$$

Vecteurs de forces nodales équivalentes aux :

- $\{F\}_s$ densité surfacique d'efforts imposés $\underline{t}_{\text{imp}}$
- $\{F\}_{\text{reac}}$ densité surfacique de réaction aux déplacements imposés $\underline{u}_{\text{imp}}$
- $\{F\}_v$ densité volumique d'efforts imposés $\underline{f}_{\text{imp}}$
- $[B]\{\sigma\}$ densité volumique d'efforts intérieurs

Matrices :

- $[N]$ matrice des fonctions de forme
- $[B]$ matrice des dérivées des fonctions de forme

Mécanique

Rappels

- Hypothèse des petites déformations : $\{\varepsilon\} = [B]\{U\}$
- Élasticité linéaire : $\{\sigma\} = [C]\{\varepsilon\}$

Formulation faible et discrétisée de l'équilibre :


$$\underbrace{\int_V [B]^T [C] [B] dV}_{[K]} \{U\} = \{F\}$$

Matrices :

- $[C]$ matrice de Hooke
 $[K]$ matrice de rigidité

6 Problème étudié

- Élasticité linéaire




$$E = 200 \text{ GPa}$$


$$\nu = 0.3$$

$$\alpha = 10^{-5} \text{ K}^{-1}$$

- Déplacements imposés



- Forces imposées



6 Mécanique linéaire

Élasticité

- Objectif : calcul mécanique linéaire
en déplacements et forces imposées

$$[K]\{U\} = \{F\} \quad \Rightarrow \quad \text{Système linéaire}$$

- Méthode :
 - calcul de la matrice de rigidité
 - calcul des chargement nodaux équivalents
 - résolution avec **RESO**

$$\begin{aligned} &[K] \\ &\{F\} \\ &\{U\} = [K]^{-1}\{F\} \end{aligned}$$

6 Mécanique linéaire

Élasticité

- Restitution des objets (maillage, paramètres, ...)

```
OPTI 'REST' 'formation_debutant_2_thermique.sauv' ;  
REST ;
```

- Nouveaux paramètres

```
** Parametres materiau / material parameters  
you      = 200.E9 ;  
nu       = 0.3 ;  
alpha    = 1.E-5 ;  
  
** Parametres des conditions limites et chargement /  
* Parameters for boundary conditions and loading  
mass1    = 2500. ;
```

6 Mécanique linéaire

Élasticité

- Formulation mathématique (élasticité)

```
** Modele mecanique elastique / Elastic mechanical model
mom1 = MODE vol1 'MECANIQUE' 'ELASTIQUE' ;
mam1 = MATE mom1 'YOUN' you 'NU' nu
        'ALPH' alpha 'TREF' t0 'TALP' t0 ;
```

6 Mécanique linéaire

Élasticité

- Formulation mathématique (élasticité)

```
** Modele mecanique elastique / Elastic mechanical model
mom1 = MODE vol1 'MECANIQUE' 'ELASTIQUE' ;
mam1 = MATE mom1 'YOUN' you 'NU' nu
      'ALPH' alpha 'TREF' t0 'TALP' t0 ;
```

- Matrice de rigidité

```
** Matrice de rigidite / Stifness matix
rig = RIGI mom1 mam1 ;
```

$$[K] = \int_V [B]^T [C] [B] dV$$

6 Mécanique linéaire

Élasticité

- Formulation mathématique (élasticité)

```
** Modele mecanique elastique / Elastic mechanical model  
mom1 = MODE vol1 'MECANIQUE' 'ELASTIQUE' ;  
mam1 = MATE mom1 'YOUN' you 'NU' nu  
        'ALPH' alpha 'TREF' t0 'TALP' t0 ;
```

- Matrice de rigidité

```
** Matrice de rigidite / Stiffness matrix  
rig = RIGI mom1 mam1 ;
```

$$[K] = \int_V [B]^T [C] [B] dV$$

- Conditions aux limites

```
** Matrice de blocage / Boundary condition matrix  
blm = BLOQ 'UX' 'UY' 'UZ' sut1 ;
```

6 Mécanique linéaire

Élasticité

- Comment représenter le chargement ?

6 Mécanique linéaire


Élasticité

- Comment représenter le chargement ?

- 1) Avec une **force ponctuelle**

☞ Récupérer le "centre" de la surface (voir : POIN 'PROC')

☞ Appliquer une force ponctuelle (voir : FORC)



6 Mécanique linéaire

Élasticité


- Comment représenter le chargement ?

- 1) Avec une **force ponctuelle**

☞ Récupérer le "centre" de la surface (voir : *POIN 'PROC'*)

☞ Appliquer une force ponctuelle (voir : *FORC*)

```
ptmass = sint POIN 'PROCHE' (1 (e / 2.) ((h / 2.) - r)) ;  
fmass1 = FORC (0. 0. (mass1 * -9.81)) ptmass ;
```



6 Mécanique linéaire

Élasticité



- Comment représenter le chargement ?

- 1) Avec une **force ponctuelle**

☞ Récupérer le "centre" de la surface (voir : POIN 'PROC')

☞ Appliquer une force ponctuelle (voir : FORC)

```
ptmass = sint POIN 'PROCHE' (l (e / 2.) ((h / 2.) - r)) ;  
fmass1 = FORC (0. 0. (mass1 * -9.81)) ptmass ;  
vecf1 = VECT fmass1 'FORC' 'ROUG' 1.E-6 ;  
TRAC vecf1 (are1 ET sint) ;
```



6 Mécanique linéaire


Élasticité

- Comment représenter le chargement ?

- 2) Avec une **pression**

☞ Récupérer "la moitié" de *sint* (voir : COOR/POIN/ELEM)

☞ Appliquer une pression (voir : PRES 'MASS')




6 Mécanique linéaire

Élasticité

- Comment représenter le chargement ?

- 2) Avec une **pression**



☞ Récupérer "la moitié" de *sint* (voir : COOR/POIN/ELEM)


☞ Appliquer une pression (voir : PRES 'MASS')

```
zsint  = COOR 3 sint ;  
ptmass = zsint POIN 'EGINFE' (h / 2.) ;  
elmass = sint ELEM 'APPU' 'STRI' ptmass ;  
pres1  = mass1 * 9.81 / (MESU elmass) ;  
fmass2 = PRES 'MASS' mom1 pres1 elmass ;
```

6 Mécanique linéaire


Élasticité

- Comment représenter le chargement ?
- 2) Avec une **pression**



- ☞ Récupérer "la moitié" de *sint* (voir : COOR/POIN/ELEM)
- ☞ Appliquer une pression (voir : PRES 'MASS')

```
zsint  = COOR 3 sint ;  
ptmass = zsint POIN 'EGINFE' (h / 2.) ;  
elmass = sint ELEM 'APPU' 'STRI' ptmass ;  
pres1  = mass1 * 9.81 / (MESU elmass) ;  
fmass2 = PRES 'MASS' mom1 pres1 elmass ;  
vecf2  = VECT fmass2 'FORC' 'VERT' 3.E-5 ;  
TRAC vecf2 (are1 ET sint) ;
```




6 Mécanique linéaire

Élasticité

- Comment représenter le chargement ?
- 3) Avec une **force surfacique**


👉 Appliquer une force surfacique (voir : FSUR 'MASS')



6 Mécanique linéaire

Élasticité

- Comment représenter le chargement ?
- 3) Avec une **force surfacique**




👉 Appliquer une force surfacique (voir : FSUR 'MASS')

```
fmass3 = FSUR 'MASS' mom1 (0. 0. (-1. * pres1)) elmass ;
```

6 Mécanique linéaire

Élasticité

- Comment représenter le chargement ?
- 3) Avec une **force surfacique**




👉 Appliquer une force surfacique (voir : FSUR 'MASS')

```
fmass3 = FSUR 'MASS' mom1 (0. 0. (-1. * pres1)) elmass ;
```

```
vecf3 = VECT fmass3 'FORC' 'TURQ' 3.E-5 ;
```

```
TRAC vecf3 (are1 ET sint) ;
```



6 Mécanique linéaire

Élasticité

- Résolution des 3 cas de chargement

```
** Resolution des 3 cas de chargement /  
* Solving the 3 load cases  
u61 u62 u63 = RESO (rig ET blm) fmass1 fmass2 fmass3 ;
```


6 Mécanique linéaire

Élasticité

- Résolution des 3 cas de chargement

```
** Resolution des 3 cas de chargement /  
* Solving the 3 load cases  
u61 u62 u63 = RESO (rig ET blm) fmass1 fmass2 fmass3 ;
```

- Tracé des déformées

```
** Deformees / Deformed shapes  
def61 = DEFO vol1 u61 50. 'ROUG' ;  
def62 = DEFO vol1 u62 50. 'VERT' ;  
def63 = DEFO vol1 u63 50. 'TURQ' ;
```

6 Mécanique linéaire




Élasticité

- Résolution des 3 cas de chargement

```
** Resolution des 3 cas de chargement /  
* Solving the 3 load cases  
u61 u62 u63 = RESO (rig ET blm) fmass1 fmass2 fmass3 ;
```

- Tracé des déformées

```
** Deformees / Deformed shapes  
def61 = DEFO vol1 u61 50. 'ROUG' ;  
def62 = DEFO vol1 u62 50. 'VERT' ;  
def63 = DEFO vol1 u63 50. 'TURQ' ;  
  
TRAC 'CACH' def61 ;  
TRAC 'CACH' def62 ;  
TRAC 'CACH' def63 ;
```

AMPLITUDE
50.AMPLITUDE
50.

6 Mécanique linéaire


Élasticité

- Résolution des 3 cas de chargement

```
** Resolution des 3 cas de chargement /  
* Solving the 3 load cases  
u61 u62 u63 = RESO (rig ET blm) fmass1 fmass2 fmass3 ;
```

- Tracé des déformées

```
** Deformees / Deformed shapes  
def61 = DEFO vol1 u61 50. 'ROUG' ;  
def62 = DEFO vol1 u62 50. 'VERT' ;  
def63 = DEFO vol1 u63 50. 'TURQ' ;  
  
TRAC 'CACH' def61 ;  
TRAC 'CACH' def62 ;  
TRAC 'CACH' def63 ;  
  
def61a = DEFO are1 u61 50. 'ROUG' ;  
def62a = DEFO are1 u62 50. 'VERT' ;  
def63a = DEFO are1 u63 50. 'TURQ' ;  
  
TRAC (def61a ET def62a ET def63a) ;
```



AMPLITUDE
50.
50.
50.

6 Mécanique linéaire

Élasticité


- Forces de réaction

```
** Forces de reactions / Reaction forces
```

```
rea63 = REAC u63 blm ;
```

```
vrea63 = VECT rea63 'FORC' 'LIME' ;
```

```
TRAC vrea63 are1 ;
```



FX FY FZ

6 Mécanique linéaire


Élasticité

- Forces de réaction


```
** Forces de reactions / Reaction forces  
rea63 = REAC u63 blm ;  
vrea63 = VECT rea63 'FORC' 'LIME' ;  
  
TRAC vrea63 are1 ;
```

- Déformations et contraintes

```
** Deformations / Strains  
eps63 = EPSI u63 mom1 'LINE' ;  
  
** Contraintes / Stresses  
sig63 = ELAS mom1 eps63 mam1 ;  
  
TRAC sig63 mom1 ;
```



FX FY FZ



VONMISES
< 2.43E+08
> 1.05E+06
2.39E+08
2.27E+08
2.16E+08
2.05E+08
1.93E+08
1.82E+08
1.71E+08
1.60E+08
1.48E+08
1.37E+08
1.26E+08
1.14E+08
1.03E+08
9.16E+07
8.03E+07
6.90E+07
5.76E+07
4.63E+07
3.50E+07
2.37E+07
1.24E+07

6 Mécanique linéaire


Élasticité

- Forces de réaction


```
** Forces de reactions / Reaction forces  
rea63 = REAC u63 blm ;  
vrea63 = VECT rea63 'FORC' 'LIME' ;  
  
TRAC vrea63 are1 ;
```

- Déformations et contraintes

```
** Deformations / Strains  
eps63 = EPSI u63 mom1 'LINE' ;  
  
** Contraintes / Stresses  
sig63 = ELAS mom1 eps63 mam1 ;  
  
TRAC sig63 mom1 ;  
  
def63 = DEFO vol1 u63 50. ;  
  
TRAC sig63 mom1 def63 are1 ;
```



FX FY FZ



VONMISES
< 2.43E+08
> 1.05E+06

2.39E+08
2.27E+08
2.16E+08
2.05E+08
1.93E+08
1.82E+08
1.71E+08
1.60E+08
1.48E+08
1.37E+08
1.26E+08
1.14E+08
1.03E+08
9.16E+07
8.03E+07
6.90E+07
5.76E+07
4.63E+07
3.50E+07
2.37E+07
1.24E+07

AMPLITUDE
oooooooooooo

Nouvel objet : DEFORMEE

6 Mécanique linéaire

Élasticité


- Contrainte de von Mises

**** Contraintes aux points de Gauss / Stresses at Gauss points**

vm63 = VMIS mom1 sig63 ;

MESS 'Mini et Maxi de von Mises : ' (**MINI** vm63) (**MAXI** vm63) ;

TRAC vm63 mom1 ;



6 Mécanique linéaire

Élasticité

- Contrainte de von Mises

**** Contraintes aux points de Gauss / Stresses at Gauss points**



vm63 = VMIS mom1 sig63 ;

MESS 'Mini et Maxi de von Mises : ' (MINI vm63) (MAXI vm63) ;

TRAC vm63 mom1 ;

vm63g = CHAN 'CHPO' vm63 mom1 'SUPP' ;

TRAC vm63g ;




6 Mécanique linéaire

Élasticité

- Évolution d'un champ le long d'une ligne du maillage

```
** Valeurs le long d'un bord /  
* Values along an edge  
vm63p = CHAN 'CHPO' vm63 mom1 ;  
evs1   = EVOL 'CHPO' vm63p 115 ;  
  
DESS evs1 ;
```



6 Mécanique linéaire

Élasticité

- Évolution d'un champ le long d'une ligne du maillage

```

** Valeurs le long d'un bord /
* Values along an edge
vm63p = CHAN 'CHPO' vm63 mom1 ;
evs1   = EVOL 'CHPO' vm63p 115 ;

```

```
DESS evs1 ;
```


- Évolution d'un champ le long d'une ligne quelconque

```

** Valeurs le long d'une ligne quelconque /
* Values along any line
lx      = (PROG 0.01 'PAS' 0.01 0.9) * 1 ;
nx      = DIME lx ;
ly      = (PROG nx * 0.) + (e / 2.) ;
lz      = (0.4 * h * (SIN (lx * 3000.))) + (h / 2.) ;
lig1    = QUEL 'SEG2' lx ly lz COUL 'ROSE' ;

TRAC (are1 ET lig1) ;

```



6 Mécanique linéaire

Élasticité


- Évolution d'un champ le long d'une ligne du maillage

```

** Valeurs le long d'un bord /
* Values along an edge
vm63p = CHAN 'CHPO' vm63 mom1 ;
evs1   = EVOL 'CHPO' vm63p 115 ;

```

```
DESS evs1 ;
```



- Évolution d'un champ le long d'une ligne quelconque

```

** Valeurs le long d'une ligne quelconque /
* Values along any line
lx      = (PROG 0.01 'PAS' 0.01 0.9) * 1 ;
nx      = DIME lx ;
ly      = (PROG nx * 0.) + (e / 2.) ;
lz      = (0.4 * h * (SIN (lx * 3000.))) + (h / 2.) ;
lig1    = QUEL 'SEG2' lx ly lz COUL 'ROSE' ;

```


```
TRAC (are1 ET lig1) ;
```

```

vm63n = CHAN 'NOEUD' mom1 vm63 ;
vm63l = PROI vm63n lig1 ;
evs3   = EVOL 'ROSE' 'CHPO' vm63l lig1 ;


```

```
DESS evs3 ;
```




7 Problème étudié


- Élasticité linéaire
- Déplacements imposés



- Forces imposées



- Chargement thermique



7 Mécanique linéaire

Élasticité, chargement thermique

- Prise en compte de la dilatation thermique

$$\int_V [B]^T \{\sigma\} dV = \{F\}$$

$$\int_V [B]^T [C] \{\varepsilon - \varepsilon_{th}\} dV = \{F\}$$

$$\int_V [B]^T [C] \{\varepsilon\} dV = \{F\} + \int_V [B]^T [C] \{\varepsilon\}_{th} dV$$

$$[K] \{U\} = \{F\} + \underbrace{\int_V [B]^T [C] \{\varepsilon\}_{th} dV}_{\{F\}_{th}}$$

7 Mécanique linéaire

Élasticité, chargement thermique

- Objectif : calcul mécanique linéaire précédent
avec un chargement thermique supplémentaire

$$[K]\{U\} = \{F\} + \underbrace{\int_V [B]^T [C] \{\varepsilon\}_{th} dV}_{\{F\}_{th}}$$

- Méthode :
 - déformations thermiques
 - contraintes thermiques
 - forces nodales éq.
 - résolution avec **RESO**

$$\begin{aligned}\{\varepsilon\}_{th} \\ \{\sigma\}_{th} &= [C]\{\varepsilon\}_{th} \\ \{F\}_{th} &= \int_V [B]^T \{\sigma\}_{th} dV \\ \{U\} &= [K]^{-1}(\{F\} + \{F\}_{th})\end{aligned}$$

7 Mécanique linéaire

Élasticité, chargement thermique

- Déformation thermique

```
** Increment de deformations thermiques /  
* Thermal strain increment  
temp1 = tab1 . 'TEMPERATURES' . (n1 - 1) ;  
epsth = EPTH temp1 mom1 mam1 ;
```

$$\{\varepsilon\}_{\text{th}} = \alpha \{\Delta T\}$$

- Pseudo contraintes thermique

```
** Pseudo contraintes / Pseudo stresses  
sigth = ELAS mom1 epsth mam1 ;
```

$$\{\sigma\}_{\text{th}} = [C] \{\varepsilon\}_{\text{th}}$$

- Forces nodale équivalentes

```
** Forces nodales eq. dilatation /  
* Eq. nodal forces thermal expansion  
fth = BSIG mom1 sigth ;
```


$$\{F\}_{\text{th}} = \int_V [B]^T \{\sigma\}_{\text{th}} dV$$

7 Mécanique linéaire

Élasticité, chargement thermique

- Résolution et affichage des résultats

```
** Resolution / Solving  
u7      = RESO (rig ET blm) (fmass3 ET fth) ;  
  
** Deformee / Deformed shape  
def7a = DEFO are1 u7 50. 'ORAN' ;  
def0a = DEFO are1 u7 0.  'GRIS' ;  
  
TRAC (def0a ET def7a ET def63a) ;
```



7 Mécanique linéaire

Élasticité, chargement thermique

- Résolution et affichage des résultats

```
** Resolution / Solving
```

```
u7      = RESO (rig ET blm) (fmass3 ET fth) ;
```

```
** Deformee / Deformed shape
```

```
def7a = DEFO are1 u7 50. 'ORAN' ;
```

```
def0a = DEFO are1 u7 0. 'GRIS' ;
```

```
TRAC (def0a ET def7a ET def63a) ;
```



```
** Contraintes / Stresses
```

```
epstot = EPSI 'LINE' mom1 u7 ;
```

```
epsel  = epstot - epsth ;
```

```
sig7   = ELAS mom1 epsel mam1 ;
```

```
TRAC sig7 mom1 are1 ;
```




AMPLITUDE


50.
20.
0.0VONMISES
< 4.59E+08
> 3.03E+064.52E+08
4.30E+08
4.09E+08
3.88E+08
3.66E+08
3.45E+08
3.23E+08
3.02E+08
2.81E+08
2.59E+08
2.38E+08
2.17E+08
1.95E+08
1.74E+08
1.53E+08
1.31E+08
1.10E+08
8.85E+07
6.71E+07
4.57E+07
2.44E+07

8 Problème étudié


- Élasticité linéaire
- Déplacements imposés



- Forces imposées



- Chargement thermique



8 Mécanique linéaire

Élasticité, chargement thermique, matériau hétérogène

- Coefficient de dilatation α dépendant de x (loi normale)

$$\alpha(x) = \alpha_0 \left(1 + 3e^{-\left(\frac{x-\mu}{\sigma}\right)^2} \right)$$

avec :

$$\begin{aligned} \mu &= \frac{l}{2} \text{ moyenne} \\ \sigma &= \frac{l}{5} \text{ écart type} \end{aligned}$$

8 Mécanique linéaire

Élasticité, chargement thermique, matériau hétérogène

- Coefficient de dilatation α dépendant de x (loi normale)

$$\alpha(x) = \alpha_0 \left(1 + 3e^{-\left(\frac{x-\mu}{\sigma}\right)^2} \right)$$

avec :

$$\begin{aligned} \mu &= \frac{l}{2} \text{ moyenne} \\ \sigma &= \frac{l}{5} \text{ écart type} \end{aligned}$$

```
** Champ alpha(x) / Field alpha(x)
x      = COOR 1 mom1 ;
xmoy   = 1 / 2. ;
xsig   = 1 / 5. ;
```

8 Mécanique linéaire

Élasticité, chargement thermique, matériau hétérogène

- Coefficient de dilatation α dépendant de x (loi normale)

$$\alpha(x) = \alpha_0 \left(1 + 3e^{-\left(\frac{x-\mu}{\sigma}\right)^2} \right)$$

avec :

$$\mu = \frac{l}{2} \text{ moyenne}$$

$$\sigma = \frac{l}{5} \text{ écart type}$$

```
** Champ alpha(x) / Field alpha(x)
```

```
x      = COOR 1 mom1 ;
```


```
xmoy = 1 / 2. ;
```

```
xsig = 1 / 5. ;
```

```
* Champ par points / Point field
```

```
chealpha = alpha * (1. + (3. *  
                      (EXP (-1. * (((x - xmoy) / xsig) ** 2)))))) ;
```

```
TRAC chealpha mom1 are1 ;
```



8 Mécanique linéaire

Élasticité, chargement thermique, matériau hétérogène

- Mise à jour des caractéristiques matériau

- * Mise à jour des paramètres matériau /

- * Updating the material parameters

```
mam1b      = MATE mom1 'YOUN' you 'NU' nu  
              'ALPH' chealpha 'TREF' t0 'TALP' t0 ;
```

8 Mécanique linéaire

Élasticité, chargement thermique, matériau hétérogène

- Mise à jour des caractéristiques matériau

```
* Mise a jour des parmetres materiau /  
* Updating the material parameters  
mam1b      = MATE mom1 'YOUN' you 'NU' nu  
              'ALPH' chealpha 'TREF' t0 'TALP' t0 ;
```

- Mise à jour du chargement thermique


```
** Mise a jour chargement thermique / Updating thermal load  
epsth = EPTH temp1 mom1 mam1b ;  
sigth = ELAS mom1 epsth mam1b ;  
fth   = BSIG mom1 sigth ;
```

8 Mécanique linéaire

Élasticité, chargement thermique, matériau hétérogène

- Résolution et affichage des résultats

```
** Resolution / Solving  
u8      = RESO (rig ET blm) (fmass3 ET fth) ;  
  
** Deformee / Deformed shape  
def8a = DEFO are1 u8 50. 'ROUG' ;  
  
TRAC (def0a ET def7a ET def8a) ;
```



AMPLITUDE
50.
50.
0.0

8 Mécanique linéaire

Élasticité, chargement thermique, matériau hétérogène

- Résolution et affichage des résultats

```
** Resolution / Solving
```

```
u8      = RESO (rig ET blm) (fmass3 ET fth) ;
```

```
** Deformee / Deformed shape
```

```
def8a = DEFO are1 u8 50. 'ROUG' ;
```

```
TRAC (def0a ET def7a ET def8a) ;
```



```
** Contraintes / Stresses
```

```
epstot = EPSI 'LINE' mom1 u8 ;
```

```
epsel  = epstot - epsth ;
```


```
sig8   = ELAS mom1 epsel mam1b ;
```

```
TRAC sig8 mom1 are1 ;
```



9.1 Problème étudié

- Élasto plasticité (plasticité parfaite)




$$E = 200 \text{ GPa}$$

$$\nu = 0.3$$

$$\alpha(x) = 10^{-5} \left(1 + 3e^{-\left(\frac{x-\mu}{\sigma}\right)^2} \right) \text{ K}^{-1}$$

$$\underline{\sigma_y = 250 \text{ MPa}}$$



- Déplacements imposés
- Forces imposées
- Chargement thermique

9.1 Mécanique non linéaire

Élastoplasticité, chargement thermique, matériau hétérogène

- Objectif : calcul mécanique non linéaire (matériau)

$$\int_V [B]^T \{\sigma\} dV = \{F\}$$

- Méthode :
 - ajout d'un modèle de plasticité
 - description temporelle des chargements (pseudo temps)
 - résolution avec la procédure **PASAPAS**

9.1 Mécanique non linéaire

Élastoplasticité, chargement thermique, matériau hétérogène

- Formulation mathématique (élasticité)

```
sigy      = 250.E6 ;
```

```
** Nouveau modele mecanique / New mechanical model  
mom2      = MODE vol1 'MECANIQUE' 'ELASTIQUE' 'PLASTIQUE' 'PARFAIT' ;  
chealpha  = CHAN 'CHAM' chpalph mom2 ;  
mam2      = MATE mom2 'YOUN' you 'NU' nu 'SIGY' sigy  
           'ALPH' chealpha 'TREF' t0 'TALP' t0 ;
```

9.1 Mécanique non linéaire

Élastoplasticité, chargement thermique, matériau hétérogène

- Formulation mathématique (élasticité)

```
sigy      = 250.E6 ;
```

```
** Nouveau modele mecanique / New mechanical model
mom2      = MODE vol1 'MECANIQUE' 'ELASTIQUE' 'PLASTIQUE' 'PARFAIT' ;
chealpha  = CHAN 'CHAM' chpalpha mom2 ;
mam2      = MATE mom2 'YOUN' you 'NU' nu 'SIGY' sigy
           'ALPH' chealpha 'TREF' t0 'TALP' t0 ;
```

- Chargements : description temporelle

```
* 1) Forces imposees / Imposed forces
evm1      = EVOL 'MANU' (PROG 0. 100.) (PROG 0. 1.) ;
chamass   = CHAR 'MECA' fmass3 evm1 ;
```

```
* 2) Temperatures imposees / Imposed temperatures
ttemps    = TABL ;
ttemps . 0 = 0. ;
ttemps . 1 = 100. ;
ttimep    = TABL ;
ttimep . 0 = tab1 . 'TEMPERATURES' . 0 ;
ttimep . 1 = tab1 . 'TEMPERATURES' . (n1 - 1) ;
chatemp    = CHAR 'T' ttemps ttimep ;
```

9.1 Mécanique non linéaire

Élastoplasticité, chargement thermique, matériau hétérogène

- Résolution avec **PASAPAS**
👉 *Remplir la TABLE et lancer PASAPAS*

9.1 Mécanique non linéaire

Élastoplasticité, chargement thermique, matériau hétérogène

- Résolution avec **PASAPAS**

👉 *Remplir la TABLE et lancer PASAPAS*

```
** Resolution avec la procedure PASAPAS /  
* Solving with the PASAPAS procedure  
tab2                                = TABL ;  
tab2 . 'MODELE'                     = mom2 ;  
tab2 . 'CARACTERISTIQUES'           = mam2 ;  
tab2 . 'BLOCAGES_MECANIQUES'        = blm ;  
tab2 . 'CHARGEMENT'                 = chamass ET chatemp ;  
tab2 . 'TEMPS_CALCULES'              = PROG 0. 'PAS' 5. 100. ;  
PASAPAS tab2 ;
```

9.1 Mécanique non linéaire

Élastoplasticité, chargement thermique, matériau hétérogène

- Post traitement : boucle de tracés


```

n2      = DIME (tab2 . 'DEPLACEMENTS') ;
liso2 = PROG 0. 'PAS' (sigy / 20.) sigy ;
REPE b2 n2 ;
  ui     = tab2 . 'DEPLACEMENTS' . (&b2 - 1) ;
  defi   = DEFO vol1 ui 50 ;
  sigi   = tab2 . 'CONTRAINTES' . (&b2 - 1) ;
  titi   = CHAI '[9.1] Contraintes au pas' ' ' (&b2 - 1) ' / ' ' ' (n2 - 1) ;

  TRAC sigi mom2 defi are1 liso2 'TITR' titi ;

FIN b2 ;

```



VONMISES
< 2.90E+08
> 2.36E+06

2.50E+08
2.38E+08
2.25E+08
2.12E+08
2.00E+08
1.88E+08
1.75E+08
1.62E+08
1.50E+08
1.38E+08
1.25E+08
1.12E+08
1.00E+08
8.75E+07
7.50E+07
6.25E+07
5.00E+07
3.75E+07
2.50E+07
1.25E+07
0.0

AMPLITUDE
min:0.000000
max:2.36E+06


[9.1] Contraintes au pas 20 / 20

9.1 Mécanique non linéaire

Élastoplasticité, chargement thermique, matériau hétérogène


- Post traitement : boucle de tracés

```
emax = MAXI (EXCO 'EPSE' (tab2 . 'VARIABLES_INTERNES' . (n2 - 1))) ;  
liso3 = PROG 0. 'PAS' (emax / 20.) emax ;  
REPE b2 n2 ;  
  ui    = tab2 . 'DEPLACEMENTS' . (&b2 - 1) ;  
  defi  = DEFO vol1 ui 50 ;  
  epi   = tab2 . 'VARIABLES_INTERNES' . (&b2 - 1) ;  
  titi  = CHAI '[9.1] Variables internes au pas' ' ' (&b2 - 1) ' / ' ' (n2 - 1) ;  
  
  TRAC epi mom2 defi are1 liso3 'TITR' titi ;  
  
FIN b2 ;
```



9.2 Problème étudié

- Élasto plasticité (plasticité parfaite), dépendance à la température



$$E = 200 \text{ GPa}$$

$$\nu = 0.3$$

$$\alpha(x) = 10^{-5} \left(1 + 3e^{-\left(\frac{x-\mu}{\sigma}\right)^2} \right) \text{ K}^{-1}$$

$$\underline{\sigma_y = f(T)}$$

- Déplacements imposés
- Forces imposées
- Chargement thermique

9.2 Mécanique non linéaire

Élastoplasticité, chargement thermique, matériau hétérogène, T dépendant

- Dépendance d'un paramètre à la température


```
** Evolution de SIGY selon T /  
* Evolution of SIGY with T  
evsigy = EVOL 'MANU'  
'T'      (PROG 0. 100. 150. 200. 215. 230. 235. 240. 250. 275. 300.)  
'SIGY' ((PROG 1. 0.99 0.98 0.94 0.9 0.8 0.4 0.33 0.28 0.23 0.20) * sigy) ;
```

9.2 Mécanique non linéaire

Élastoplasticité, chargement thermique, matériau hétérogène, T dépendant

- Dépendance d'un paramètre à la température

```
** Evolution de SIGY selon T /  
* Evolution of SIGY with T  
evsigy = EVOL 'MANU'  
'T'      (PROG 0. 100. 150. 200. 215. 230. 235. 240. 250. 275. 300.)  
'SIGY' ((PROG 1. 0.99 0.98 0.94 0.9 0.8 0.4 0.33 0.28 0.23 0.20) * sigy) ;  
  
DESS evsigy ;
```



9.2 Mécanique non linéaire

Élastoplasticité, chargement thermique, matériau hétérogène, T dépendant

- Dépendance d'un paramètre à la température

```

** Evolution de SIGY selon T /
* Evolution of SIGY with T
evsigy = EVOL 'MANU'
'T'      (PROG 0. 100. 150. 200. 215. 230. 235. 240. 250. 275. 300.)
'SIGY' ((PROG 1. 0.99 0.98 0.94 0.9 0.8 0.4 0.33 0.28 0.23 0.20) * sigy) ;

DESS evsigy ;


```

- Mise à jour des caractéristiques matériau

```

* Mise a jour des parametres materiau /
* Updating the material parameters
mam2 = MATE mom2 'YOUN' you 'NU' nu 'SIGY' evsigy
      'ALPH' chealpha 'TREF' t0 'TALP' t0 ;

```



9.2 Mécanique non linéaire

Élastoplasticité, chargement thermique, matériau hétérogène, T dépendant

- Résolution avec PASAPAS

```
** Resolution avec la procedure PASAPAS /  
* Solving with the PASAPAS procedure  
tab2                                = TABL ;  
tab2 . 'MODELE'                     = mom2 ;  
tab2 . 'CARACTERISTIQUES'          = mam2 ;  
tab2 . 'BLOCAGES_MECANIQUES'       = blm ;  
tab2 . 'CHARGEMENT'                 = chamass ET chatemp ;  
tab2 . 'TEMPS_CALCULES'             = PROG 0. 'PAS' 5. 100. ;  
PASAPAS tab2 ;
```

9.2 Mécanique non linéaire

Élastoplasticité, chargement thermique, matériau hétérogène, T dépendant

- Post traitement : boucle de tracés



```
n2 = DIME (tab2 . 'DEPLACEMENTS') ;

emax = MAXI (EXCO 'EPSE' (tab2 . 'VARIABLES_INTERNES' . (n2 - 1))) ;
liso3 = PROG 0. 'PAS' (emax / 20.) emax ;
REPE b2 n2 ;
  ui = tab2 . 'DEPLACEMENTS' . (&b2 - 1) ;
  defi = DEFO vol1 ui 50 ;
  epi = tab2 . 'VARIABLES_INTERNES' . (&b2 - 1) ;
  titi = CHAI '[9.2] Variables internes au pas' ' ' (&b2 - 1) ' / ' ' (n2 - 1) ;

  TRAC epi mom2 defi are1 liso3 'TITR' titi ;

FIN b2 ;

FIN ;
```



AMPLITUDE
normale

Fichiers solution

- Les fichiers solution de cette formation sont des cas tests téléchargeables sur le site web :

<https://www-cast3m.cea.fr/index.php?page=exemples>

⇒ formation_debutant_1_maillage.dgibi

⇒ formation_debutant_2_thermique.dgibi

⇒ formation_debutant_3_mecanique.dgibi

Éléments finis barre/poutre, coque, joints...

- Le choix des éléments finis se fait dans **MODE**

```
mod1 = MODE lig1 'MECANIQUE' 'ELASTIQUE' | 'POUT' | ;  
                                           | 'TIMO' |
```

```
mod1 = MODE sur1 'MECANIQUE' 'ELASTIQUE' | 'DKT' | ;  
                                           | 'COQ4' |
```

- Les caractéristiques géométriques sont données dans **MATE**

```
mat1 = MATE mod1 'YOUN' 210.E9 'NU' 0.3 'SECT' 1.E-2  
                'INRY' 1.E-4 'INRZ' 2.E-4 'TORS' 3.E-4 ;
```

```
mat1 = MATE mod1 'YOUN' 210.E9 'NU' 0.3 'EPAI' 1.E-2 ;
```

- On peut alors agir sur les d.d.l. de déplacement et de rotation

```
b1 = BLOQ p1 'UX' 'UY' 'UZ' 'RX' 'RY' 'RZ' ;
```

Éléments finis barres en 3D


- Exemple d'un treillis de barres type grue (en 3D)

```
OPTI 'DIME' 3 'ELEM' 'SEG2' ;
p1 = 0. 0. 0. ;
p2 = 0. 3. 0. ;
p3 = 0. 1.5 3. ;
p123 = p1 ET p2 ET p3 ;
mail = QUEL 'SEG2' p1 p2 p3 p1 ;
REPE bou 6 ;
  p4 p5 p6 = p1 p2 p3 PLUS (3. 0. 0.) ;
  mail = mail ET (QUEL 'SEG2' p1 p4 p5 p2 p4 p6 p5 p3 p6 p1) ;
  p1 = p4 ;
  p2 = p5 ;
  p3 = p6 ;
FIN bou ;

mo = MODE mail 'MECANIQUE' 'BARR' ;
ma = MATE mo 'YOUN' 210.E9 'NU' 0.3 'SECT' 2.E-3 ;
ri = RIGI mo ma ;

b1 = BLOQ 'DEPL' p123 ;
f = FORC (0. 0. -1.E4) (p4 ET p5) ;
vf = VECT f 'FORC' 'ROUG' 1.E-3 ;
TRAC (ri ET b1) vf ;


u = RESO (ri ET b1) f ;
df = DEFO mail u 300. ;
sig = SIGM mo ma u ;
TRAC sig mo df ;
```



Éléments finis coques en 3D

- Exemple d'une coque cylindrique sous pression (en 3D)

```
OPTI 'DIME' 3 'ELEM' 'QUA4' ;  
p1 = 4. 0. 0. ;  
lig = CERC 45 'ROTA' 360. p1 (0. 0. 0.) (0. 0. 1.) ;  
ELIM lig 1.E-2 ;  
sur = lig TRAN 20 (0. 0. 10.) ;  
  
mo = MODE sur 'MECANIQUE' 'COQ4' ;  
ma = MATE mo 'YOUN' 210.E9 'NU' 0.3 'EPAI' 0.3 ;  
ri = RIGI mo ma ;  
  
bl = BLOQ 'UX' 'UY' 'UZ' 'RX' 'RY' 'RZ' lig ;  
f = PRES 'COQU' mo 42.E5 'NORM' ;  
  
u = RESO (ri ET bl) f ;  
df = DEFO sur u 700. ;  
TRAC 'CACH' df ;
```



Éléments finis coques en 2D

- Le même cas en 2D (axisymétrique)

```
OPTI 'DIME' 2 'MODE' 'AXIS' 'ELEM' 'SEG2' ;  
p1 = 4. 0. ;  
lig = DROI 20 p1 (4. 10.) ;  
  
mo = MODE lig 'MECANIQUE' 'COQ2' ;  
ma = MATE mo 'YOUN' 210.E9 'NU' 0.3 'EPAI' 0.3 ;  
ri = RIGI mo ma ;  
  
bl = BLOQ 'UR' 'UZ' 'RT' p1 ;  
f = PRES 'COQU' mo -42.E5 'NORM' ;  
  
u = RESO (ri ET bl) f ;  
df = DEFO lig u 700. ;  
TRAC df ;
```

Et d'autres options (2D, 1D, ...)

- Exemple d'un problème de conduction stationnaire axisymétrique


```
OPTI 'DIME' 2 'MODE' 'AXIS' 'ELEM' 'QUA8' ;
```

```
p1 = 10. 0. ;  
p2 = 50. 0. ;  
lig = DROI 10 p1 p2 ;  
sur = lig TRAN 1 (0. 1.) ;  
li1 = sur COTE 4 ;  
li2 = sur COTE 2 ;
```

```
mo = MODE sur 'THERMIQUE' ;  
ma = MATE mo 'K' 42. ;  
con = COND mo ma ;
```

```
bl1 = BLOQ 'T' li1 ;  
bl2 = BLOQ 'T' li2 ;  
f1 = DEPI bl1 20. ;  
f2 = DEPI bl2 500. ;
```

```
t = RESO (con ET bl1 ET bl2) (f1 ET f2) ;  
TRAC t sur ;  
evt = EVOL 'CHPO' t 'T' lig ;  
DESS evt ;
```



Et d'autres options (2D, 1D, ...)

- Le même cas en 1D (cylindrique)


```
OPTI 'DIME' 1 'MODE' 'UNID' 'AXIS' 'ELEM' 'SEG3' ;
```

```
p1 = POIN 10. ;  
p2 = POIN 50. ;  
lig = DROI 10 p1 p2 ;
```

```
mo = MODE lig 'THERMIQUE' ;  
ma = MATE mo 'K' 42. ;  
con = COND mo ma ;
```

```
bl1 = BLOQ 'T' p1 ;  
bl2 = BLOQ 'T' p2 ;  
f1 = DEPI bl1 20. ;  
f2 = DEPI bl2 500. ;
```

```
t = RESO (con ET bl1 ET bl2) (f1 ET f2) ;  
TRAC t lig ;  
evt = EVOL 'CHPO' t 'T' lig ;  
DESS evt ;
```



Lire / Sortir des données

- **SAUV**egarder au format binaire et **REST**ituer

```
OPTI 'SAUV' 'toto' ;    SAUV ;  
OPTI 'REST' 'toto' ;    REST ;
```

Fichier XDR

- Exécuter une commande **EXTE**rieure

```
tab1 = EXTE 'grep -in mecanique toto.dgibi' 'RC' ;  
tab1 contient le résultat de la commande grep
```

- **ACQU**érir un fichier texte

Lire un fichier texte, ligne par ligne

```
OPTI 'ACQU' 'toto.txt' ;  
ACQU n1*'ENTIER' x1*'FLOTTANT' ;  
ACQU n2*'ENTIER' list1*'LISTREEL' 4 ;
```

toto.txt					
		X		Y	Z
1	3.14				
2	25.2	28.3		24.3	16.6

Lire / Sortir des données

- Écrire un fichier texte (**SORT**)

```
OPTI 'SORT' 'fibonacci.txt' ;

SORT 'CHAI' 'Hello' ;
SORT 'CHAI' 4 8 15 16 23 42 ;
SORT 'CHAI' ' ' ;

SORT 'CHAI' 'Iter. Fibo' ;
fibo1 = 0 ;
fibo2 = 1 ;
SORT 'CHAI' 0 fibo1 ;
SORT 'CHAI' 1 fibo2 ;
REPE b1 15 ;
    fibo3 = fibo2 ;
    fibo2 = fibo1 + fibo2 ;
    fibo1 = fibo3 ;
    SORT 'CHAI' (&b1 + 1) fibo2 ;
FIN b1 ;
```

Hello					
4	8	15	16	23	42
Iter.	Fibo				
0	0				
1	1				
2	1				
3	2				
4	3				
5	5				
6	8				
7	13				
8	21				
9	34				
10	55				
11	89				
12	144				
13	233				
14	377				
15	610				
16	987				

Lire / Sortir des données

- Lire/Écrire des **données tabulées** en colonnes
Fichier .csv



Objets concernés : LISTENTI, LISTREEL, LISTMOTS, EVOLUTIO, TABLE

Utilisé par les éditeurs de texte ou tableur (Excel)

```
OPTI 'SORT' 'toto.csv' ;  
SORT 'EXCE' obj1 ;  
  
tab1 = LIRE 'CSV' 'toto.csv' ;
```

Choix possible du séparateur de colonnes :

point virgule, virgule, espace, tabulation, barre oblique...

Lire / Sortir des données

- Écrire des données au **format VTK**

Fichier .vtk

Objets concernés : MAILLAGE, CHPOINT, MCHAML

Utilisé par Paraview

```
OPTI 'SORT' 'toto.vtk' ;  
SORT 'VTK' mail1 dep1 sig1 ;
```



- Lire/Écrire des données au **format MED**

Fichier .med

Objets concernés : MAILLAGE, CHPOINT, MCHAML, TABLE

Utilisé par Salomé, EuroPlexus, Code Aster

```
OPTI 'SORT' 'toto.med' ;  
SORT 'MED' mail1 dep1 sig1 ;  
  
tab1 = LIRE 'MED' 'toto.med' ;
```



Développement : procédures Gibiane

- Écrire chaque procédure dans un fichier texte :
 - extension ".procedur"
 - nom de fichier = nom de procédure
 - 1 fichier = 1 procédure
- Emplacements possibles :
 - ./
 - ./procedur
- Lancer Cast3M comme d'habitude
castem24 toto.dgibi
- Idem pour les notices (fichiers ".notice")

Développement : sources Esope

- L'utilisateur peut modifier/corriger/ajouter le code source des opérateurs et directives
- Compilation des fichiers source Esope
`compilcast24 toto.eso titi.eso ...`
- Édition des liens
`essaicast24`
⇒ création d'un fichier exécutable binaire : `cast_24`
⇒ version locale de Cast3M
- Lancer Cast3M comme d'habitude
`castem24 toto.dgibi`

Et pour finir

- Consulter la **documentation** régulièrement
~ 70 instructions découvertes durant cette formation
près de 1400 instructions existantes !
- Inscription à la **liste de diffusion Cast3M** (voir le site web Cast3M)
Envoyer un e-mail vide à `sympa@umontpellier.fr`
avec comme objet du message :

```
SUB cast3m-util last_name first_name
```


et rien d'autre ! (pas de message, pas de signature, ...)
- **Club Cast3M** : séminaire annuel des utilisateurs
Chaque année en novembre dans le sud de Paris
Présentation de travaux réalisés avec Cast3M, nouveautés de la prochaine version
Inscription gratuite !